LOOP Center, Inc.

$2.00

DIGITAL EQUIPMENT CORPORATION

education

HUNTINGTON I Application Programs – MATHEMATICS



computers are for kids — digital — EduSystems—expandable, economical

DEC EDUCATIONAL PUBLICATIONS

Other publications in the continuing series of curriculum material published by DEC for use with EduSystems 10 through 80 are listed below. Please inquire directly for prices on classroom quantities.

Additional publications may be obtained from:

Software Distribution Center
Digital Equipment Corporation
Maynard, Massachusetts 01754

Populution: Self-teaching BASIC Workbook	$2.00
BASIC Application Programs	
Mathematics I	1.00
Mathematics II	1.00
Science	1.00
Business/Social Studies	1.00
Plotting	1.00
BASIC Matrix Operations	1.00
Huntington I Application Programs	
MATHEMATICS	2.00
TEACHER ASSISTANCE	1.00
Huntington I Simulation Programs	
BIOLOGY	1.00
CHEMISTRY	2.00
EARTH SCIENCE	1.00
PHYSICS	2.00
SOCIAL STUDIES	1.00
Problems for Computer Mathematics	1.25

HUNTINGTON COMPUTER PROJECT

A TEACHER'S MANUAL

(COMPUTER - RELATED MATERIALS)

Second Edition

January 31, 1971

Director: Dr. Ludwig Braun
Assistant Director: Dr. Marian Visich, Jr.

Polytechnic Institute of Brooklyn
333 Jay Street
Brooklyn, New York 11201

Developed by the Huntington Computer Project during the period May, 1968 and September, 1970. This effort was supported by the National Science Foundation under Grant No. J000079.

The enclosed material is a compilation of computer programs developed during the period May, 1968 to September, 1970. These programs were developed by teachers and students in the high schools which participated with us, and by the Project staff.

All of the enclosed programs have been tested on a Digital Equipment Corporation TSS-8 time-shared computer during the summer of 1970. To the best of our ability, we have assured ourselves that the programs actually run. It should be pointed out, however, that we were not able to make an exhaustive exploration of the programs. There may be undiscovered bugs (if there aren't, it may be the first time in the history of computing). We would appreciate hearing of any which emerge in the future.

These programs run in the version of BASIC which existed on the TSS-8 in August, 1970, and should run on most other versions of BASIC. The major potential problem on other machines is the output format (DEC uses 14 columns per print zone, while some other manufacturers use 15; we used the TAB function, which doesn't exist in all BASIC compiles). It may be necessary to make some minor changes in programs to adjust this format. Another possible problem is in the use of the RANDOMIZE command in some programs to start the random-number generator at a random point. If this command is not available, some other means should be devised for randomizing the start.

It is our sincere hope that these programs and their supporting documentation will be helpful to educators who are exploring the uses of computers in education.

We are anxious to hear of any bugs, errors, or improvements in these programs, and are especially anxious to hear of any novel ways of using them.

Ludwig Braun

Marian Visich, Jr.

TABLE OF CONTENTS

MATHEMATICS

MATHEMATICS (cont.)

DISCIPLINE MATHEMATICS 9th YEAR

SUBJECT MULTIPLICATION INVOLVING ONE AND TWO DIGIT MULTIPLIERS

PROGRAM NAME ARITH

DESCRIPTION:

This program calls upon students, in a random fashion, to perform multiplication problems. Each student calculates five separate problems, and is allotted three chances to respond with the correct answer. At the conclusion of three incorrect responses, the computer will report to the student the correct answer. At the conclusion of five problems, the student will be given a score, and a new student will be called. Each factor will not exceed 100.

OBJECTIVES:

To review and reinforce students' ability to multiply by one and two digit multipliers.

PRELIMINARY PREPARATION:

None

DISCUSSION:

A. Operational Suggestions - This particular program is designed for group study, and may be used for an entire period. A portable TV was originally used to display computer output.

B. Follow-up - By modifying line numbers 520 and 560 to

```
520 LET X = A + B
560 PRINT A "+" B  " =";
```

this program becomes practice in addition. Modifications may also be made for division, subtraction and individual remedial work.

C. MODIFICATION-If your computer has string capability, student names may be used rather than student numbers, by making a few minor programming changes. This change (having the computer type out the student's name) increases the student interest.

```
HELLO CLASS.....TODAY I WANT TO REVIEW  MULTIPLICATION
WITH YOU.   WHEN I CALL ON YOU PLEASE COME UP AND TYPE IN
YOUR ANSWERS.  IF YOU ARE WRONG YOU GET TWO MORE CHANCES.
HOW MANY STUDENTS ARE THERE IN THE CLASS TODAY? 30
TEACHER!...GIVE EVERYONE A NUMBER FROM 1 TO 30


OK, STUDENT NO.  27  IT'S YOUR TURN!

GIVE ME YOUR LUCKY NUMBER
? 5


 81  X  54  = ? 4374
YOU'RE RIGHT STUDENT NO. 27


 63  X  97  = ? 6111
YOU'RE RIGHT STUDENT NO. 27


 62  X  17  = ? 1054
YOU'RE RIGHT STUDENT NO. 27


 50  X  78  = ? 3900
YOU'RE RIGHT STUDENT NO. 27


 93  X  81  = ? 7533
YOU'RE RIGHT STUDENT NO. 27
YOU GOT  5  RIGHT OUT OF 5 PROBLEMS.
GOODBYE...STUDENT NO. 27


OK, STUDENT NO.  25  IT'S YOUR TURN!

GIVE ME YOUR LUCKY NUMBER
?
↑C

READY
```

```
100 REM W. TEPPER, WYANDANCH HS, 4/21/69
105 REM  REVISED BY C.LOSIK  8-5-70
106 REM  WE DO A RANDOM PROBLEM FOR EACH STUDENT,  A -OP- B
110 REM THIS PROGRAM CALLS STUDENTS IN A RANDOM FASHION TO DO INDIVIDUAL
120 REM PROBLEMS.  BY MODIFYING A FEW STATEMENTS I CAN CHANGE THE
130 REM TYPE OF PROBLEMS.
140 REM REVISED 5/7/69
150 PRINT "HELLO CLASS.....TODAY I WANT TO REVIEW ";
151 REM  CHANGE BELOW FOR YOUR OPERATION
152 PRINT " MULTIPLICATION"
160 PRINT
170 PRINT "WITH YOU.    WHEN I CALL ON YOU PLEASE COME UP AND TYPE IN"
180 PRINT
190 PRINT "YOUR ANSWERS.  IF YOU ARE WRONG YOU GET TWO MORE CHANCES."
200 PRINT
210 PRINT "HOW MANY STUDENTS ARE THERE IN THE CLASS TODAY";
220 INPUT S
230 PRINT
240 PRINT "TEACHER!...GIVE EVERYONE A NUMBER FROM 1 TO"S
250 PRINT
260 PRINT
270 PRINT
272 REM  YOU MUST RANDOMIZE THE PROCESS FOR BEST RESULTS
275 RANDOMIZE
280 LET Q=INT(RND(-2)*S)
290 PRINT "OK, STUDENT NO. "Q" IT'S YOUR TURN!"
300 LET J=0
310 LET L =0
320 PRINT
330 PRINT
340 PRINT "GIVE ME YOUR LUCKY NUMBER"
350 INPUT Z
360 FOR T=1 TO Z
370 LET A=INT(RND(-2)*100)
380 LET B=INT(RND(-5)*100)
390 NEXT T
400 LET N=0
410 LET J=J+1
415 REM  X IS THE ANSWER TO A -OP- B
420 LET X=A*B
430 PRINT
440 PRINT
450 PRINT
455 REM  PRINT A -OP- B = ?
460 PRINT A" X "B" = ";
470 INPUT K
480 IF ABS(K-X)<.005 THEN 590
490 LET N=N+1
```

```
500 IF N=3 THEN 530
510 PRINT "YOU'RE WRONG...TRY AGAIN"
520 GO TO 460
530 PRINT "YOUR WRONG AGAIN"
540 PRINT "THE ANSWER IS "X
550 IF J<5 THEN360
560 PRINT "YOU GOT "L"  RIGHT OUT OF 20 PROBLEMS"
570 PRINT "GOOD BYE .... STUDENT NO."Q
580 GO TO 250
590 PRINT "YOU'RE RIGHT STUDENT NO."Q
600 LET L = L+1
610 IF J<5 THEN360
620 PRINT "YOU GOT "L" RIGHT OUT OF 5 PROBLEMS."
630 PRINT "GOODBYE...STUDENT NO."Q
640 GO TO 260
650 END
```

DISCIPLINE MATHEMATICS-SOCIAL SCIENCE

SUBJECT FINANCIAL PROBLEMS

PROGRAM NAME BANK

DESCRIPTION:

This program solves financial problems concerning installment buying, long-term loans, and savings accounts. The program gives you a choice of these three types of problems, and asks for the information needed to do said problem.

OBJECTIVES:

A. This program aids students in learning the terms used in certain financial problems.

B. Student will hopefully be motivated to learn the mathematical logic behind the solution of these problems.

PRELIMINARY PREPARATION:

A. Student - A review of decimals and fractions would be helpful.

B. Materials - A terminal, and a means by which to display the output to an entire class (e.g. overhead projector, closed circuit TV, etc.)

DISCUSSION:

A type of problem may be demonstrated through the use of the computer, then the mathematical logic behind the solution of the problem may be developed through the use of a flow chart similar to the one that follows.

Terminology may be taught when the computer asks for input (see sample run).

Since the execution time of one run is extremely short, many more problems may be demonstrated. Depending upon the ability of the class or student, a variety of relationships may be discovered.

FINANCIAL PROBLEMS

```
THIS PROGRAM SOLVES THREE TYPES OF PROBLEMS:

     (1) INTEREST ON INSTALLMENT BUYING
     (2) PAYMENTS ON LONG TERM LOAN
     (3) BALANCE OF A SAVINGS ACCOUNT

WHICH PROBLEM WOULD YOU LIKE TO WORK WITH (TYPE 1, 2 OR 3)? 1

*****

THIS SECTION WILL DETERMINE THE ACTUAL INTEREST YOU PAY
WHEN YOU PURCHASE SOMETHING ON CREDIT.

WHAT IS THE CASH PRICE OF THE ARTICLE ($)? 88.99
          DOWN PAYMENT ($)? 10
          NUMBER OF PAYMENTS EXCLUDING THE DOWN PAYMENT? 18
          NUMBER OF PAYMENTS PER MONTH? 1
          AMOUNT PER PAYMENT ($)? 4.85


THE RATE OF INTEREST CHARGED WAS 5.69  PERCENT.


*****

WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)? 1
WHICH PROBLEM WOULD YOU LIKE TO WORK WITH (TYPE 1, 2 OR 3)? 2

*****

THIS SECTION WILL DETERMINE PAYMENTS FOR A LONG TERM LOAN.

WHAT IS THE AMOUNT BORROWED ($)? 3000
          INTEREST CHARGED (%)? 8
          INTERVAL BETWEEN PAYMENTS (MONTHS)? 1
          TERM OF THE LOAN (YEARS)? 2

DO YOU WISH TO SEE THE TOTALS ONLY - INSTEAD OF THE ENTIRE
TABLE - (1-YES, 0-NO)? 0
```

PERIOD	OUTSTANDING PRINCIPAL AT BEGINNING OF PERIOD	INTEREST DUE AT END OF PERIOD	PRINCIPAL REPAID AT END OF PERIOD
1	3000	20	115.68
2	2884.32	19.23	116.45
3	2767.87	18.45	117.23
4	2650.64	17.67	118.01
5	2532.63	16.88	118.8
6	2413.83	16.09	119.59
7	2294.24	15.29	120.39
8	2173.85	14.49	121.19
9	2052.66	13.68	122
10	1930.66	12.87	122.81
11	1807.85	12.05	123.63
12	1684.22	11.23	124.45
13	1559.77	10.4	125.28
14	1434.49	9.56	126.12
15	1308.37	8.72	126.96
16	1181.41	7.88	127.8
17	1053.61	7.02	128.66
18	924.95	6.17	129.51
19	795.44	5.3	130.38
20	665.06	4.43	131.25
21	533.81	3.56	132.12
22	401.69	2.68	133
23	268.69	1.79	133.89
24	134.8	.9	134.78
TOTALS		256.34	3000

```
YOUR MONTHLY PAYMENT IS $ 135.68  AND TOTALS $ 3256.34
```

```
*****

WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)? 1
WHICH PROBLEM WOULD YOU LIKE TO WORK WITH (TYPE 1, 2 OR 3)? 3

*****

THIS SECTION CALCULATES THE BALANCE OF A SAVINGS ACCOUNT
IN WHICH DEPOSITS ARE MADE REGULARLY.

WHAT IS THE AMOUNT DEPOSITED PER INTEREST PERIOD ($)? 10000
HOW OFTEN IS THE INTEREST COMPOUNDED (MONTHS)? 3
WHAT IS THE RATE OF INTEREST PAID (%)? 5
FOR HOW LONG WILL YOU DEPOSIT MONEY (YEARS)? 5


THE BALANCE OF YOUR ACCOUNT AFTER  5 YEARS WILL BE $ 202500


*****

WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)? 0


READY
```

```
100 REM FINANCIAL PROBLEMS   A. WEBB  12/67
101 REM REVISED 8/25/70 (D. PESSEL)
110 PRINT TAB(20);"FINANCIAL PROBLEMS"
115 REM REVISED BY W. TEPPER,  WYANDANCH H.S.  7/10/69
120 PRINT
130 PRINT"THIS PROGRAM SOLVES THREE TYPES OF PROBLEMS:"
132 PRINT
134 PRINT"      (1) INTEREST ON INSTALLMENT BUYING"
136 PRINT"      (2) PAYMENTS ON LONG TERM LOAN"
138 PRINT"      (3) BALANCE OF A SAVINGS ACCOUNT"
140 PRINT
142 PRINT"WHICH PROBLEM WOULD YOU LIKE TO WORK WITH (TYPE 1, 2 OR 3)";
144 INPUT Q1
146 PRINT
147 PRINT"*****"
148 PRINT
150 IF Q1>2 THEN 820
155 IF Q1>1 THEN 260
160 GO TO 590
260PRINT "THIS SECTION WILL DETERMINE PAYMENTS FOR A LONG TERM LOAN."
270 PRINT
280 PRINT"WHAT IS THE AMOUNT BORROWED ($)";
281 INPUT A
285 PRINT"              INTEREST CHARGED (%)";
286 INPUT I
290 PRINT"              INTERVAL BETWEEN PAYMENTS (MONTHS)";
291 INPUT P
295 PRINT"              TERM OF THE LOAN (YEARS)";
296 INPUT Y
300 PRINT
360 PRINT"DO YOU WISH TO SEE THE TOTALS ONLY - INSTEAD OF THE ENTIRE"
361 PRINT"TABLE - (1-YES, 0-NO)";
362 INPUT P5
370 PRINT
375 IF P5>0 THEN 430
380 PRINT"            OUTSTANDING"
390 PRINT"            PRINCIPAL AT                                PRINCIPAL"
400 PRINT"            BEGINNING          INTEREST DUE AT          REPAID AT"
410 PRINT"PERIOD      OF PERIOD          END OF PERIOD            END OF PERIOD"
420 PRINT
430 LET Z=(Y*12)/P
440 LET K=(I*(P/12))/100
445 LET E=A*K/(1-1/(1+K)↑Z)
446 LET E=INT(E*100+.5)/100
450 LET C=A
460 LET F=0
461 LET D1=0
470 LET T1=0
480 LET T1=T1+1
490 IF T1>Z THEN 554
500 LET B=T1
510 LET C=C-F
520 LET D=C*K
522 LET F=E-D
525 LET C=INT(C*100+.5)/100
530 LET D=INT(D*100+.5)/100
535 LET F=INT(F*100+.5)/100
541 LET D1=D1+D
548 IF P5>0 THEN 480
550 PRINT B;TAB(11);C;TAB(29);D;TAB(48);F
```

```
552 GO TO 480
554 IF P5<1 THEN 561
555 PRINT
556 LET D1=INT(D1*100+.5)/100
558 PRINT"TOTAL INTEREST PAID - $"D1
559 PRINT"TOTAOL PRINCIPAL REPAID - $"A
560 GO TO 565
561 PRINT"                                   ---------------     -------------"
564 PRINT"TOTALS";TAB(29);D1;TAB(48);A
565 LET E5=INT((D1+A)*100+.5)/100
566 PRINT
567 LET E6=E5/((Y*12)/P)
568 LET E6=INT(100*E6+.5)/100
569 PRINT"YOUR MONTHLY PAYMENT IS $"E6" AND TOTALS $"E5
570 GO TO 1060
590 PRINT"THIS SECTION WILL DETERMINE THE ACTUAL INTEREST YOU PAY"
600 PRINT"WHEN YOU PURCHASE SOMETHING ON CREDIT."
610 PRINT
620 PRINT"WHAT IS THE CASH PRICE OF THE ARTICLE ($)";
621 INPUT C
630 PRINT"              DOWN PAYMENT ($)";
631 INPUT D
640 PRINT"              NUMBER OF PAYMENTS EXCLUDING THE DOWN PAYMENT";
641 INPUT N
650 PRINT"              NUMBER OF PAYMENTS PER MONTH";
651 INPUT S
660 PRINT"              AMOUNT PER PAYMENT ($)";
661 INPUT R
690 PRINT
720 LET B=R*N+D
730 LET I=B-C
740 LET M=N/(S*12)
750 LET T=I*100/(B*M)
760 PRINT
770 PRINT
775 LET T=INT(100*T+.5)/100
780PRINT "THE RATE OF INTEREST CHARGED WAS"T" PERCENT."
790 GO TO 1060
820 PRINT"THIS SECTION CALCULATES THE BALANCE OF A SAVINGS ACCOUNT"
830 PRINT"IN WHICH DEPOSITS ARE MADE REGULARLY."
840 PRINT
860 PRINT"WHAT IS THE AMOUNT DEPOSITED PER INTEREST PERIOD ($)";
861 INPUT A
870 PRINT"HOW OFTEN IS THE INTEREST COMPOUNDED (MONTHS)";
871 INPUT B
880 PRINT"WHAT IS THE RATE OF INTEREST PAID (%)";
881 INPUT C
890 PRINT"FOR HOW LONG WILL YOU DEPOSIT MONEY (YEARS)";
891 INPUT D
950 LET F=0
960 LET E=(C/100)/(12/B)
970 LET G=(12/B)*D
980 LET T1=0
990 LET T1=T1+1
1000 IF T1=G+1 THEN 1030
1010 LET F=(E*A)+(A+F)
1020 GO TO 990
1030 PRINT
1040 PRINT
1045 LET F=INT(100*F+.5)/100
1050 PRINT"THE BALANCE OF YOUR ACCOUNT AFTER "D"YEARS WILL BE $"F
1060 PRINT
1070 PRINT
1080 PRINT
1081 PRINT"*****"
1082 PRINT
1084 PRINT"WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)";
1086 INPUT Q4
1090 IF Q4>0 THEN 142
1100 END
```

DISCIPLINE CALCULUS-GRADE 13

SUBJECT LENGTH OF ANY CURVE

PROGRAM NAME CRVLEN

DESCRIPTION:

This program approximates the length of any curve between two fixed points on the curve, by taking an increasing number of subintervals and computing the sum of the secants involved.

OBJECTIVES:

A. Time saving factor for computations.

B. By typing out successive approximations, the machine displays the manner by which the limit is approached.

C. The attendant discussion focuses attention upon the techniques needed to build up the analytic method for finding the length of a curve.

PRELIMINARY PREPARATION: None

DISCUSSION:

The operator inserts any function, sets up his own limits, and the computer proceeds to print out several approximations to the actual length a diagram (such as below) should be displayed, indicating the geometric basis for the computations.



Piecewise linear approximation of a smooth curve

```
                    LENGTH OF A CURVE

THIS PROGRAM APPROXIMATES THE LENGTH OF ANY CURVE BETWEEN
TWO POINTS HAVING P AND Q AS THEIR RESPECTIVE ABSCISSAS.
THE PROGRAM DIVIDES THE CURVE INTO INCREASING NUMBERS OF
SUBINTERVALS, JOINS THESE WITH SECANTS AND FINDS THE SUM
OF THESE SECANTS.

TO INPUT THE FUNCTION WHICH YOUR CURVE REPRESENTS, TYPE AS
FOLLOWS AFTER THE PROGRAM STOPS:
(TYPE THE 'RETURN' KEY AFTER EACH LINE INCLUDING 'RUN')

          1 GO TO 200
          300 DEF FNY(X)=....(YOUR FUNCTION OF X)....
          RUN

FOR EXAMPLE, TO USE THE FUNCTION 2*X↑3+3*X↑2-2*X+3
YOU WOULD TYPE:

          1 GO TO 200
          300 DEF FNY(X)=2*X↑3+3*X↑2-2*X+3
          RUN

YOU MIGHT TRY THAT AS YOUR FIRST RUN.


READY

1 GO TO 200
300 DEF FNY(X)=2*X↑3+3*X↑2-2*X+3
RUN

WHAT ARE THE ABSCISSAS OF THE END POINTS OF THE INTERVAL
WHOSE LENGTH YOU WANT (SMALLER ONE FIRST:P,Q)? -1,6

 NUMBER OF         SUM OF
SUBINTERVALS   SECANT LENGTHS   % CHANGE IN LENGTH
------------   --------------   ------------------

 1               585.0467        NO PREVIOUS VALUE
 2               585.1583         .02125142
 4               529.6522         .6557383
 8               531.0171         .2576957
 16              531.9642         .1783583
 32              532.0166         9.834262E-3
 64              532.0416         4.713729E-3
 128             532.0485         1.287718E-3
 256             532.0501         3.068687E-4

*****

WOULD YOU LIKE TO TRY NEW END POINTS (1-YES, 0-NO)? 0

TO TRY ANOTHER FUNCTION, RETYPE LINE 300, AND 'RUN'.
SEE INSTRUCTIONS FOR MORE DETAILS.  IF YOU ARE FINISHED,
TYPE '1' AND 'RETURN' KEY AFTER THE PROGRAM STOPS.


READY

1
```

```
100 REM LENGTH OF A CURVE-G.J. O'CONNOR 7-29-68
101 REM REVISED 8-7-70 (D. PESSEL) (COMBINATION OF LECUQ & LEFUQ)
103 REM IMPORTANT VARIABLES:  S-SECANT LENGTH; S1-PREVIOUS SECANT
104 REM LENGTH; P-PERCENT CHANGE IN SECANT LENGTH
110 PRINT TAB(20); "LENGTH OF A CURVE"
111 PRINT
120 PRINT "THIS PROGRAM APPROXIMATES THE LENGTH OF ANY CURVE BETWEEN"
121 PRINT "TWO POINTS HAVING P AND Q AS THEIR RESPECTIVE ABSCISSAS."
122 PRINT "THE PROGRAM DIVIDES THE CURVE INTO INCREASING NUMBERS OF"
123 PRINT "SUBINTERVALS, JOINS THESE WITH SECANTS AND FINDS THE SUM"
124 PRINT "OF THESE SECANTS."
125 PRINT
126 PRINT "TO INPUT THE FUNCTION WHICH YOUR CURVE REPRESENTS, TYPE AS"
127 PRINT "FOLLOWS AFTER THE PROGRAM STOPS:"
128 PRINT "(TYPE THE 'RETURN' KEY AFTER EACH LINE INCLUDING 'RUN')"
129 PRINT
130 PRINT "          1 GO TO 200"
131 PRINT "          300 DEF FNY(X)=....(YOUR FUNCTION OF X)...."
132 PRINT "          RUN"
133 PRINT
134 PRINT "FOR EXAMPLE, TO USE THE FUNCTION 2*X↑3+3*X↑2-2*X+3"
135 PRINT "YOU WOULD TYPE:"
136 PRINT
137 PRINT "          1 GO TO 200"
138 PRINT "          300 DEF FNY(X)=2*X↑3+3*X↑2-2*X+3"
139 PRINT "          RUN"
140 PRINT
141 PRINT "YOU MIGHT TRY THAT AS YOUR FIRST RUN."
150 STOP
200 REM CALCULATION AND PRINTING IF RESULTS
205 PRINT "WHAT ARE THE ABSCISSAS OF THE END POINTS OF THE INTERVAL"
206 PRINT "WHOSE LENGTH YOU WANT (SMALLER ONE FIRST:P,Q)";
207 INPUT P,Q
208 IF P<Q THEN 213
209 PRINT "P MUST BE LESS THAN Q!"
210 GO TO 207
212 INPUT P,Q
213 PRINT
214 PRINT " NUMBER OF","    SUM OF"
215 PRINT "SUBINTERVALS","SECANT LENGTHS","  % CHANGE IN LENGTH"
216 PRINT "------------","--------------","  ------------------"
217 PRINT
230 LET S1=0
250 FOR N=1 TO 9
300 DEF FNY(X)=2*X↑3+3*X↑2-2*X+3
310 LET E=2↑(N-1)
320 LET H=(Q-P)/E
330 LET S=0
340 FOR I=0 TO E-1
350 LET L=SQR((FNY(P+I*H+H)-FNY(P+I*H))↑2+H*H)
360 LET S=S+L
370 NEXT I
372 IF S1>0 THEN 375
373 PRINT E,S,"  NO PREVIOUS VALUE"
374 GO TO 385
375 LET P5=((ABS(S1-S))/S1)*100
380 PRINT E,S,"  "P5
385 LET S1=S
390 NEXT N
400 PRINT
401 PRINT "*****"
402 PRINT
403 PRINT "WOULD YOU LIKE TO TRY NEW END POINTS (1-YES, 0-NO)";
404 INPUT Q1
405 IF Q1>0 THEN 210
410 PRINT
445 PRINT "TO TRY ANOTHER FUNCTION, RETYPE LINE 300, AND 'RUN'."
446 PRINT "SEE INSTRUCTIONS FOR MORE DETAILS.  IF YOU ARE FINISHED,"
447 PRINT "TYPE '1' AND 'RETURN' KEY AFTER THE PROGRAM STOPS."
500 END
```

DISCIPLINE CALCULUS - GRADE 13

SUBJECT AREA UNDER ANY CURVE,
(ANALYTICALLY DEFINED)

PROGRAM NAME CVAREA

DESCRIPTION:

By numerical methods, this program evaluates the definite integral of f(x), from x = a to x = b, by four different methods of successive approximation:

I Rectangles (starting with f(a) as height)

II Rectangles (starting with f(a+h) as height)

III Trapezoids

IV Parabolas (Simpson's Rule)

OBJECTIVES:

A. Enhances comprehension of the analytic procedures for finding the area under a curve.

B. Dramatizes the limiting processes involved.

C. Decreases the time needed for lengthy computations.

PRELIMINARY PREPARATION:

Prior to the computer run, diagrams should appear on the board, or on the overhead projector screen to demonstrate the geometric significance of the computer output.



Inscribed Rectangular Approximation



Circumscribed Rectangular Approximation

DISCUSSION:

This program may be run as an introduction to the problem of finding the area under a curve. In some classes, the consideration of Simpson's Rule may be omitted or briefly hinted at. With the more mathematically talented classes, an explanation of this parabolic approximation should precede the running of the program.



Trapezoid Approximation

AREA UNDER A CURVE - INTEGRATION

THIS PROGRAM EVALUATES THE DEFINITE INTEGRAL OF F(X)
FROM X=A TO X=B BY FOUR METHODS OF NUMERICAL APPROXIMATION:

I RECTANGLES (INITIAL HEIGHT OF F(X))
II RECTANGLES (INITIAL HEIGHT OF F(X+H))
III TRAPEZOIDS
IV PARABOLAS (SIMPSON'S RULE)

AFTER THE PROGRAM STOPS, YOU MAY ENTER YOUR FUNCTION AS FOLLOWS:

```
1 GO TO 200
300 DEF FNY(X)=...(YOUR FUNCTION OF X)...
RUN
```

FOR EXAMPLE, TO FIND THE AREA UNDER THE CURVE Y=X↑3 YOU
WOULD TYPE:

```
1 GO TO 200
300 DEF FNY(X)=X↑3
RUN
```

YOU MIGHT TRY THAT AS YOUR FIRST RUN.
END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY.

READY

```
1 GO TO 200
300 DEF FNY(X)=X↑3
RUN
```

WHAT ARE YOUR VALUES FOR A AND B (SMALLER FIRST:A,B)? 1,10

NUMBER OF SUBINTERVALS	I. SUM OF RECTANGLES	II. SUM OF RECTANGLES	III. SUM OF TRAPEZOIDS	IV. SUM OF PARABOLAS
2	753.1875	5248.687	3000.937	2499.75
4	1501.172	3748.922	2625.047	2499.75
8	1969.137	3093.012	2531.074	2499.75
16	2226.612	2788.55	2507.581	2499.75
32	2361.223	2642.192	2501.708	2499.75
64	2429.997	2570.481	2500.239	2499.75

NOTE THAT SIMPSON'S RULE (IV) CONVERGES FASTEST.

WOULD YOU LIKE TO TRY NEW VALUES FOR A AND B (1-YES, 0-NO)? 0

*****

TO USE A NEW FUNCTION YOU NEED ONLY RETYPE LINE 300
AND 'RUN'. SEE INSTRUCTIONS FOR MORE DETAILS.
IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY.

READY

1

```
100 REM AREA UNDER A CURVE-Q.J. O'CONNOR
101 REM REVISED 8/18/70 (D. PESSEL) (COMBINATION OF DEFIN AND ACCUQ)
102 REM IMPORTANT VARIABLES: D-# OF SUBINTERVALS; AREA BY
103 REM RECTANGLES (F(X))-P, BY RECTANGLES (F(X+H))-Q,
104 REM BY TRAPEZOIDS-T, BY PARABOLAS-S; C-STORES PREVIOUS
105 REM VALUE OF S.
110 PRINT TAB(15); "AREA UNDER A CURVE - INTEGRATION"
111 PRINT
112 PRINT "     THIS PROGRAM EVALUATES THE DEFINITE INTEGRAL OF F(X)"
113 PRINT "FROM X=A TO X=B BY FOUR METHODS OF NUMERICAL APPROXIMATION:"
114 PRINT
115 PRINT TAB(20); "I RECTANGLES (INITIAL HEIGHT OF F(X))"
116 PRINT TAB(19); "II RECTANGLES (INITIAL HEIGHT OF F(X+H))"
117 PRINT TAB(18); "III TRAPEZOIDS"
118 PRINT TAB(19); "IV PARABOLAS (SIMPSON'S RULE)"
119 PRINT
120 PRINT "AFTER THE PROGRAM STOPS, YOU MAY ENTER YOUR FUNCTION AS";
121 PRINT " FOLLOWS:"
122 PRINT
123 PRINT TAB(13); "1 GO TO 200"
124 PRINT TAB(13); "300 DEF FNY(X)=...(YOUR FUNCTION OF X),.."
125 PRINT TAB(13); "RUN"
126 PRINT
127 PRINT "FOR EXAMPLE, TO FIND THE AREA UNDER THE CURVE Y=X↑3 YOU"
128 PRINT "WOULD TYPE:"
129 PRINT
130 PRINT TAB(13);"1 GO TO 200"
131 PRINT TAB(13);"300 DEF FNY(X)=X↑3"
132 PRINT TAB(13);"RUN"
133 PRINT
134 PRINT "YOU MIGHT TRY THAT AS YOUR FIRST RUN."
135 PRINT "END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY."
150 STOP
200 PRINT "WHAT ARE YOUR VALUES FOR A AND B (SMALLER FIRST:A,B)";
201 INPUT A,B
203 IF B>=A THEN 210
204 PRINT "A MUST BE LESS THAN B!"
205 GO TO 200
210 PRINT
211 PRINT " NUMBER OF","I. SUM OF","II. SUM OF","III. SUM OF",
212 PRINT "IV. SUM OF"
213 PRINT "SUBINTERVALS","RECTANGLES","RECTANGLES","TRAPEZOIDS",
214 PRINT "PARABOLAS"
218 PRINT "------------","----------","----------","----------",
219 PRINT "----------"
250 LET M=-2
260 LET S=0
300 DEF FNY(X)=X↑3
310 LET M=M+3
320 FOR N=M TO M+2
330 LET C=S
340 LET Q=0
350 LET P=0
360 LET D=2↑N
```

```
365 PRINT D,
370 LET H=(B-A)/D
380 FOR I=0 TO (D-1)
390 LET P=P+H*FNY(A+I*H)
400 LET Q=Q+H*FNY(A+I*H+H)
410 NEXT I
415 PRINT P,Q,
420 LET T=(P+Q)/2
425 PRINT T,
430 LET U=FNY(A)+FNY(B)
440 FOR J=2 TO (D-2) STEP 2
450 LET U=U+2*FNY(A+J*H)
460 NEXT J
470 LET V=0
480 FOR K=1 TO (D-1) STEP 2
490 LET V=V+4*FNY(A+K*H)
500 NEXT K
510 LET S=(U+V)*(H/3)
520 PRINT S
530 NEXT N
535 IF D<64 THEN 310
540 IF ABS((C-S)/((C+S)/2))>.0001 THEN 310
550 PRINT
560 PRINT "NOTE THAT SIMPSON'S RULE (IV) CONVERGES FASTEST."
600 PRINT
610 PRINT "WOULD YOU LIKE TO TRY NEW VALUES FOR A AND B (1-YES, 0-NO)";
611 INPUT Q5
612 PRINT
613 PRINT "*****"
614 PRINT
615 IF Q5>0 THEN 200
620 PRINT "TO USE A NEW FUNCTION YOU NEED ONLY RETYPE LINE 300"
621 PRINT "AND 'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS."
622 PRINT "IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY."
650 END
```

DISCIPLINE MATHEMATICS, JR. HIGH

GENERAL MATH

SUBJECT GREATEST COMMON DIVISOR

PROGRAM NAME GCD

DESCRIPTION:

This program finds the greatest common divisor for two or more numbers.

OBJECTIVES:

To aid the teacher in demonstrating a method of finding the greatest common divisor.

PRELIMINARY PREPARATION:

See discussion.

DISCUSSION:

It is suggested that the teacher explain the meaning of the greatest common divisor prior to using this program, and show a number of examples.

By using the flow chart which follows, the method and logic the computer uses, can be explained to students. It is suggested that a supplementary device be used to display output to class-size groups.

```
THIS PROGRAM WILL FIND THE GREATEST COMMON DIVISOR
FOR TWO OR MORE NUMBERS.
HOW MANY NUMBERS DO YOU WISH TO INVESTIGATE? 3
TYPE IN THE NUMBERS, ONE AFTER EACH QUESTION MARK.
? 12
? 36
? 96
THE NUMBERS 12  36  96 HAVE THE G.C.D.  12


ANOTHER SET OF NUMBERS (1=YES, 0=NO) ? 1
HOW MANY NUMBERS DO YOU WISH TO INVESTIGATE? 3
TYPE IN THE NUMBERS, ONE AFTER EACH QUESTION MARK.
? 20
? 36
? 96
THE NUMBERS 20  36  96 HAVE THE G.C.D.  4


ANOTHER SET OF NUMBERS (1=YES, 0=NO) ? 1
HOW MANY NUMBERS DO YOU WISH TO INVESTIGATE? 3
TYPE IN THE NUMBERS, ONE AFTER EACH QUESTION MARK.
? 20
? 36
? 97
THE NUMBERS 20  36  97 ARE RELATIVELY PRIME.


ANOTHER SET OF NUMBERS (1=YES, 0=NO) ? 0


READY
```

```
100 REM W. TEPPER  WYANDANCH H.S.    -   MATHEMATICS
110 REM  REVISED BY C.LOSIK  8-10-70
111 REM  X(I) ARE THE NUMBERS (UP TO 100)
120 PRINT "THIS PROGRAM WILL FIND THE GREATEST COMMON DIVISOR"
130 PRINT "FOR TWO OR MORE NUMBERS."
140 DIM X(100)
150 PRINT "HOW MANY NUMBERS DO YOU WISH TO INVESTIGATE";
160 INPUT N
165 IF ABS(N-INT(N))<.0001 THEN 170
166 PRINT "TRY AGAIN."
167 GO TO 150
170 PRINT "TYPE IN THE NUMBERS, ONE AFTER EACH QUESTION MARK."
175 LET S=1E25
180 FOR K=1 TO N
190 INPUT X(K)
193 IF X(K)>S THEN 210
200 LET S=X(K)
210 NEXT K
220 LET G=0
230 FOR M=2 TO S
240 FOR I=1 TO N
250 IF X(I)/M<>INT(X(I)/M) THEN 300
260 NEXT I
290 LET G=M
300 NEXT M
310 PRINT "THE NUMBERS";
320 FOR T=1 TO N
330 PRINT X(T);
340 NEXT T
350 IF G>0 THEN 380
360 PRINT "ARE RELATIVELY PRIME."
370 GO TO 390
380 PRINT "HAVE THE G.C.D. ";G
390 PRINT
400 PRINT
410 PRINT "ANOTHER SET OF NUMBERS (1=YES, 0=NO) ";
420 INPUT Z
430 IF Z=1 THEN 150
440 IF Z=0 THEN 470
450 PRINT "TYPE 1 OR 0 AS DIRECTED."
460 GO TO 420
470 END
```

DISCIPLINE CALCULUS - GRADE 13

SUBJECT LIMIT OF $\frac{\sin x}{x}$

PROGRAM NAME LIMSIN

DESCRIPTION:

This program demonstrates that the limit of $\frac{\sin x}{x}$, as x approaches 0, equals 1, provided x is measured in radians. If x is measured in degrees, the limit equals approximately .017.

OBJECTIVES:

A. To demonstrate the manner by which the limit of $\frac{\sin x}{x}$ is approached.

B. To show that degree measure does not yield the same solution as radian measure.

PRELIMINARY PREPARATION:

A. Student

Knowledge of degree vs. radian measure.

B. Materials

None

DISCUSSION:

Following the computer type-out, the teacher will use the analytic method to evaluate the limit. Prior to this discussion, the student should be reminded of the area formulas for a triangle and for a sector in terms of the central angle measured in radians. A geometric diagram should be presented showing the sector lying between two triangles.



Here, $\frac{1}{2}r^2 \sin\theta \leq \frac{1}{2}r^2\theta \leq \frac{1}{2}r^2\tan\theta$

Circular Sector with Circumscribed and Inscribed Triangles

The teacher can modify the type-out by inserting: 195 Go to 300. This decreases the number of lines typed out to the final eleven appearing on the program "run".

THIS PROGRAM DEMONSTRATES THAT THE LIMIT OF
F(X) = (SIN X)/X, AS X APPROACHES 0, IS EQUAL TO 1,
PROVIDED X IS MEASURED IN RADIANS.

$$\lim_{X \to 0} \frac{\sin(X)}{X} = 1$$

WHEN X IS IN DEGREES,		WHEN X IS IN RADIANS,	
X IS	F(X) IS	X IS	F(X) IS
90	.01111111	1.570795	.6366203
85	.01171994	1.483529	.6715035
80	.01231009	1.396262	.705317
75	.01287901	1.308996	.7379134
70	.01342418	1.221729	.7691492
65	.01394319	1.134463	.7988866
60	.01443375	1.047197	.8269936
55	.01489367	.9599303	.8533449
50	.01532088	.8726639	.8778225
45	.01571347	.7853975	.9003165
40	.01606968	.6981311	.9207256
35	.01638789	.6108647	.9389575
30	.01666665	.5235983	.9549297
25	.01690472	.4363319	.9685698
20	.01710099	.3490656	.9798156
15	.01725459	.2617992	.988616
10	.0173648	.1745328	.9949308
5	.01743113	.08726639	.9987313
1	.01745239	.01745328	.9999492
.9	.01745256	.01570795	.9999589
.8	.01745271	.01396262	.9999675
.7	.01745284	.01221729	.9999751
.6	.01745296	.01047197	.9999817
.5	.01745306	8.726639E-3	.9999873
.4	.01745314	6.981312E-3	.9999919
.3	.0174532	5.235984E-3	.9999954
.2	.01745324	3.490656E-3	.999998
.1	.01745327	1.745328E-3	.9999995
.09	.01745327	1.570795E-3	.9999996
.08	.01745327	1.396262E-3	.9999997
.07	.01745327	1.221729E-3	.9999998
.06	.01745327	1.047197E-3	.9999998
.05	.01745328	8.726639E-4	.9999999
.04	.01745328	6.981311E-4	.9999999
.03	.01745328	5.235984E-4	1
.02	.01745328	3.490656E-4	1
.01	.01745328	1.745328E-4	1

READY

```
100 REM  BRUCE BRENT   HHHH   BKLYN POLY   7/11/69
105 REM  REVISED BY C.LOSIK   8-27-70
110 PRINT "     THIS PROGRAM DEMONSTRATES THAT THE LIMIT OF"
115 PRINT "F(X) = (SIN X)/X,  AS X APPROACHES 0, IS EQUAL TO 1,"
117 PRINT "PROVIDED X IS MEASURED IN RADIANS."
120 PRINT
125 PRINT " ","        SIN(X)"
130 PRINT " ","LIMIT   ------  =  1"
135 PRINT " ","X-->0     X"
140 PRINT
150 PRINT
160 PRINT "WHEN X IS IN DEGREES,","WHEN X IS IN RADIANS,"
165 PRINT "----------------------","----------------------"
170 PRINT "X IS","F(X) IS","X IS","F(X) IS"
175 PRINT "----","-------","----","-------"
180 PRINT
200 FOR Y=90 TO 5 STEP -5
210 LET Z=Y
220 LET Z=3.14159*Z/180
230 LET X=SIN(Z)/Z
240 LET Q=SIN(Z)/Y
250 PRINT Y,Q,Z,X
260 NEXT Y
270 PRINT
300 FOR Y=1 TO .1 STEP -.1
310 LET Z=Y
320 LET Z=3.14159*Z/180
330 LET X=SIN(Z)/Z
340 LET Q=SIN(Z)/Y
350 PRINT Y,Q,Z,X
360 NEXT Y
370 PRINT
400 FOR Y=.09 TO .01 STEP -.01
410 LET Z=Y
420 LET Z=3.14159*Z/180
430 LET X=SIN(Z)/Z
440 LET Q=SIN(Z)/Y
450 PRINT Y,Q,Z,X
460 NEXT Y
500 END
```

DISCIPLINE MATHEMATICS 10th YEAR GEOMETRY

SUBJECT AREA OF A CIRCLE

PROGRAM NAME PI2

DESCRIPTION:

This program computes the area of a circle and "pi" by using the areas of inscribed and circumscribed regular polygons.

OBJECTIVES:

As an introduction to the limit process and a method for approximating "pi".

PRELIMINARY PREPARATION:

A. Student - Students must know how to calculate the area of a circle and a triangle using the formulas: $A = \pi R^2$ and $A = \frac{1}{2}bh$.

B. Materials - chalkboard, board compass, and straight edge.

DISCUSSION:

Ask students to find the area of a circle without using the formula. The instructor may suggest to the class to inscribe and/or circumscribe an equilateral triangle. Have students compare the area of their figures to that of the circle. Some students will suggest to increase the number of sides and the instructor should suggest that a regular hexagon be used for convenience of drawing. This can be illustrated on the chalkboard for the class. Another comparison is made between the areas and then the students will observe that to obtain any satisfactory results, the number of sides must increase greatly. At this moment the instructor should introduce this program and explain to the class that the program will increase the number of sides of a regular polygon and compute the area of each new figure. A table is printed giving the areas of both inscribed and circumscribed regular polygons and also the number of sides for each area. The students can readily see that the machine has eliminated the tedious calculations. Now, have the students calculate the area of the circle using the formula and make a comparison of results; thus, the students can observe that the areas of the polygons approach the area of the circle.

DISCUSSION: (con't)

If students had taken a unit circle, they would have observed a method for approximating "pi".

Due to machine operation, the value of "pi" was used to convert degrees into radians. To avoid any circular reasoning, the instructor can use half-angle formulas to eliminate "pi" from this program.

```
AREA OF A CIRCLE USING INSCRIBED AND CIRCUMSCRIBED REGULAR POLYGONS

*****

WHAT IS THE RADIUS OF THE CIRCLE? 10

 INSCRIBED        CIRCUMSCRIBED     NUMBER OF     INSCRIBED       CIRCUMSCRIBED
   AREA                AREA          SIDES       % ERROR           % ERROR

 129.9039          519.6142          3           -58.65           65.4
 259.8075          346.4098          6           -17.3            10.27
 299.9998          321.5387          12          -4.51            2.35

HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE
THE AREA OF THIS CIRCLE? 100

 313.9523          314.2624          100         -.07             .03

WOULD YOU LIKE TO TRY ANOTHER NUMBER OF SIDES (1-YES, 0-NO)? 1
HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE
THE AREA OF THIS CIRCLE? 1E6
THAT MANY SIDES IS VALID, BUT NOT NECESSARY FOR A
GOOD APPROXIMATION.  USE 10000 AS THE MAXIMUM NUMBER.
HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE
THE AREA OF THIS CIRCLE? 10000

 314.159           314.159           10000       0                0

WOULD YOU LIKE TO TRY ANOTHER NUMBER OF SIDES (1-YES, 0-NO)? 0
WOULD YOU LIKE TO TRY ANOTHER RADIUS (1-YES, 0-NO)? 1

*****
```

```
WHAT IS THE RADIUS OF THE CIRCLE? 1000000
ANY RADIUS WILL WORK, BUT USE A NUMBER LESS THAN 1000.
WHAT IS THE RADIUS OF THE CIRCLE? 999

 INSCRIBED        CIRCUMSCRIBED     NUMBER OF     INSCRIBED       CIRCUMSCRIBED
   AREA               AREA            SIDES       % ERROR          % ERROR

 1.296443E+6      5.185754E+6         3           -58.65           65.4
 2.592881E+6      3.457173E+6         6           -17.3            10.27
 2.994001E+6      3.008960E+6         12          -4.51            2.35

HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE
THE AREA OF THIS CIRCLE? 10000

 3.135310E+6      3.135310E+6        10000 ,       0                0

WOULD YOU LIKE TO TRY ANOTHER NUMBER OF SIDES (1-YES, 0-NO)? 0
WOULD YOU LIKE TO TRY ANOTHER RADIUS (1-YES, 0-NO)? 0

*****


READY


100 REM ILLUSTRATION OF LIMITS USING CIRCLES AND POLYGONS
101 REM REVISED 6/3/70 (D. PESSEL)
105 REM IMPORTANT VARIABLES: A1-INSCRIBED AREA;  A2-CIRCUMSCRIBED
106 REM AREA;  A3-ACTUAL AREA;  P1-% ERROR OF A1;  P2-% ERROR OF A2
110 PRINT "AREA OF A CIRCLE USING INSCRIBED AND CIRCUMSCRIBED ";
111 PRINT "REGULAR POLYGONS"
112 PRINT
115 PRINT "*****"
116 PRINT
120 PRINT "WHAT IS THE RADIUS OF THE CIRCLE";
125 INPUT R
```

```
127 IF R<1000 THEN 131
128 PRINT "ANY RADIUS WILL WORK, BUT USE A NUMBER LESS THAN 1000."
129 GO TO 120
130 INPUT R
131 IF R>=.1 THEN 134
132 PRINT "RADIUS SHOULD BE AT LEAST .111"
133 GO TO 120
134 LET A3=3.1416*R*R
135 PRINT
136 PRINT
140 PRINT "INSCRIBED","CIRCUMSCRIBED","  NUMBER OF",
141 PRINT "INSCRIBED","CIRCUMSCRIBED"
150 PRINT "  AREA","      AREA","    SIDES"," % ERROR","   % ERROR"
155 PRINT
160 FOR K=0 TO 2
170 LET N=3*(2^K)
175 GOSUB 180
177 NEXT K
178 GO TO 240
179 REM COMPUTATION SUBROUTINE (LINES 180-220)
180 LET L=2*R*SIN(3.14159/N)
190 LET A1=R*COS(3.14159/N)*N*L/2
200 LET A2=N*(R^2)*TAN(3.14159/N)
205 LET P1=((A1-A3)/A3)*100
206 LET P2=((A2-A3)/A3)*100
210 PRINT A1,A2,"   ";N,INT(P1*100+.5)/100, INT(P2*100+.5)/100
220 RETURN
240 PRINT
250 PRINT
260 PRINT "HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE"
261 PRINT "THE AREA OF THIS CIRCLE";
262 INPUT N
263 IF N>1E5 THEN 266
264 IF N<3 THEN 266
265 GO TO 273
266 PRINT "THE NUMBER OF SIDES SHOULD BE AT LEAST THREE!!"
267 GO TO 260
268 PRINT "THAT MANY SIDES IS VALID, BUT NOT NECESSARY FOR A"
269 PRINT "GOOD APPROXIMATION.  USE 10000 AS THE MAXIMUM NUMBER."
270 GO TO 260
273 PRINT
274 LET N=INT(N+.5)
275 GOSUB 180
280 PRINT
285 PRINT "WOULD YOU LIKE TO TRY ANOTHER NUMBER OF SIDES";
286 PRINT " (1-YES, 0-NO)";
287 INPUT Q1
288 IF Q1>0 THEN 260
290 PRINT "WOULD YOU LIKE TO TRY ANOTHER RADIUS (1-YES, 0-NO)";
291 INPUT Q2
293 PRINT
294 PRINT "*****"
295 PRINT
296 IF Q2>0 THEN 120
300 END
```

DISCIPLINE MATHEMATICS 9, 10, 11, 12, 13

SUBJECT PLOTTING A GRAPH

PROGRAM NAME PLOTTR

DESCRIPTION:

This program plots the graph of any function (analytically defined) which the operator inputs into the program.

OBJECTIVES:

A. To check a student' s plotting procedures.

B. To obtain a quick plot of an involved function.

PRELIMINARY PREPARATION:

A. Student - Knowledge of coordinates, and plotting procedures.

B. Materials - graph paper for plotting

DISCUSSION:

The operator inputs any analytic function, along with the lower and the upper limits for x and the interval to appear on the x-axis.

The type-out positions x-values on the vertical axis, and y-values on the horizontal axis.

The points typed out may be connected by a smooth curve, and the graph may be rotated 90° to give the usual positioning of a function of x.

It should be noted that because the carriage spacing is discrete, many smooth curves may appear slightly jagged.

In the third sample run, a plot is made of a rather complex transcendental function. It is worth mentioning that this plot is obtained as easily, using this program as is that of the function Y=X.

The teacher should notice also, that, in this third sample run, we have found two of the roots of the function

Y=X+LOG(2*(SIN(X)) 2)-1.5xCOS(X)

(at X=1 and X=2.98). This program may be used for finding the roots of such difficult funcions.

```
    THIS PROGRAM WILL GRAPH A FUNCTION OF X BETWEEN ANY LIMITS
(A AND B) YOU CHOOSE, WITH AN INTERVAL OF YOUR CHOICE (I)
BETWEEN SUCCESSIVE VALUES OF X; IF YOU TYPE THE FOLLOWING:
        1 GO TO 220
        220 DEF FNY(X)=...(YOUR FUNCTION OF X)...
        230 LET A=...(YOUR SMALLER LIMIT OF X)...
        240 LET B=...(YOUR LARGER LIMIT OF X)...
        250 LET I=...(YOUR X-INCREMENT)...
        RUN

READY

1 GO TO 220
220 DEF FNY(X)=SIN(X)
230 LET A=0
240 LET B=7
250 LET I=.5
RUN

X
-
V
A
L
U
E
S
                        - - - - Y-VALUES - - - -
                -3              -1.5             0              1.5
                 I---------I----+----I---------I---------I----+----I
 0                                               *
 .5                                              I    *
 1                                               I       *
 1.5                                             I          *
 2                                               I       *
 2.5                                             I     *
 3                                               I*
 3.5                                          *  I
 4                                            *  I
 4.5                                       *     I
 5                                         *     I
 5.5                                          *  I
 6                                               I  *
 6.5                                             I *
 7                                               I     *

NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:
-3 ,-2 ,-1 , 0 , 1 , 2

READY

220 DEF FNY(X)=X
230 LET A=-1
240 LET B=9
250 LET I=1
RUN
```



NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:
-1 , 1 , 3 , 5 , 7 , 9

READY

1

```
1 GO TO 220
220 DEF FNY(X)=X+LOG(2*(SIN(X))↑2)-1.5*COS(X/2)
230 LET A=0.5
240 LET B=3.1
250LET I=0.1
RUN
```



NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:
-3 ,-2 ,-1 , 0 , 1 , 2

```
READY

1 GO TO 220
220 DEF FNY(X)=LOG((SIN(X))↑2)
230 LET A=1
240 LET B=3
250 LET I=0.1
RUN

X
-
V
A
L
U
E
S                    - - - - Y-VALUES - - - -
               -5              -3.5             -2              -.5
                I---------I----+----I---------I---------I----+----I
 1                                                              *  I
 1.1                                                            * I
 1.2                                                           *I
 1.3                                                           *I
 1.4                                                           *
 1.5                                                           *
 1.6                                                           *
 1.7                                                           *
 1.8                                                           *I
 1.9                                                           *I
 2                                                            * I
 2.1                                                         *  I
 2.2                                                       *    I
 2.3                                                     *      I
 2.4                                                   *        I
 2.5                                                 *          I
 2.6                                               *            I
 2.7                                             *              I
 2.8                                           *                I
 2.9                                       *                    I
 3                              *                                I

NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:
-5 ,-4 ,-3 ,-2 ,-1 , 0


READY
```

```
110 REM QUENTIN J. O'CONNOR, COMMACK H.S.NORTH, REVISED JULY,1969
115 REM  REVISED BY C.LOSIK  8-7-70
116 REM  A,B,I ARE SELF-EXPLANATORY
117 REM AUTOMATIC SCALING AND A FLOATING AXIS ARE USED.
120PRINT"     THIS PROGRAM WILL GRAPH A FUNCTION OF X BETWEEN ANY LIMITS"
130PRINT"(A AND B) YOU CHOOSE, WITH AN INTERVAL OF YOUR CHOICE (I)"
140PRINT"BETWEEN SUCCESSIVE VALUES OF X, IF YOU TYPE THE FOLLOWING:"
150 PRINT"          1 GO TO 220"
160 PRINT"          220 DEF FNY(X)=...(YOUR FUNCTION OF X)..."
170 PRINT"          230 LET A=...(YOUR SMALLER LIMIT OF X)..."
180 PRINT"          240 LET B=...(YOUR LARGER LIMIT OF X)..."
190 PRINT"          250 LET I=...(YOUR X-INCREMENT)..."
200 PRINT"          RUN"
210 STOP
220 DEF FNY(X)=X
230 LET A=-I
240 LET B=9
242 IF A<B THEN 250
244 PRINT "YOUR 'A' MUST BE LESS THAN YOUR 'B'."
246 STOP
250 LET I=1
260 LET L=FNY(A)
270 LET U=FNY(A)
280 FOR X=A TO B STEP I
290 LET Y=FNY(X)
300 IF Y-L<0 THEN 320
310GOTO 330
320 GOSUB350
330 IF Y-U>0 THEN 350
340 GO TO 360
350 GOSUB 400
360 NEXT X
370 GO TO 420
380 LET L=Y
390 RETURN
400 LET U=Y
410 RETURN
420 IF INT(U)-U=0 THEN 450
430 LET U1=INT(U)+1
440 GO TO 460
450 LET U1=U
460 LET L1=INT(L)
470 LET D=U1-L1
480 IF INT(D/5)-D/5=0 THEN 570
490 FOR K=1 TO2
500 LET L1=L1-1
510 LET D=U1-L1
520 IF INT(D/5)-D/5=0 THEN 570
530 LET U1=U1+1
540 LET D=U1-L1
550 IF INT(D/5)-D/5=0 THEN 570
560 NEXT K
```

```
570 LET E=D/5
580 PRINT "X"
590 PRINT "-"
600 PRINT "V"
610 PRINT "A"
620 PRINT "L"
630 PRINT "U"
640 PRINT "E"
650 PRINT "S                          - - - - Y-VALUES - - - -"
660PRINT"                  "L1," "jL1+E*1.5," "jL1+E*3,"    "jL1+E*4.5
670PRINT"                   I---------I----+----I---------I---------I";

671 PRINT "-----+----I"
680 IF L1>0 THEN 1020
690  LET Q=INT((-L1)*(10/E)+.5)
700 DEF FNP(Y)=INT((Y-L1)*(10/E)+.5)
710FOR X=A TO B STEP I
720 PRINTX,
730 LET Y=FNY(X)
740 IF Y>=0 THEN 850
760 PRINT TAB(15+FNP(Y));"*";TAB(Q+15);"I"
840 GO TO 1000
850 IF Y>0 THEN 910
860 PRINT TAB(Q+15);"*"
900 GO TO 1000
910 PRINT TAB(15+Q);"I";TAB(15+FNP(Y));"*"
1000 NEXT X
1010 GO TO 1100
1020 FOR X=A TO B STEP I
1030 PRINTX,
1040 LET Y= FNY(X)
1050 PRINT TAB(INT((Y-L1)*(10/E)+.5)+14);"*"
1090 NEXT X
1100 PRINT
1110 PRINT "NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:"
1120 PRINT L1;","jL1+E;","jL1+E*2;","jL1+E*3;","jL1+E*4;","jL1+E*5
1240 END
```

DISCIPLINE MATHEMATICS, GEN. 9th YR.

SUBJECT PRIME FACTOR

PROGRAM NAME PRIFA

DESCRIPTION:

This program finds the prime factors of any given integer, or prints " is prime" if the integer has no proper divisors.

OBJECTIVES:

A. To display to the student the prime factors of a large number of integers, giving the students a chance to discover relationships.

B. To use the motivation of the computer to teach the method that the program uses to find the prime factors.

PRELIMINARY PREPARATION:

A. Student - Should understand the meaning of composite, prime, factor, and prime factor.

B. Materials - If you desire to use this program with a group, a means by which the output can be displayed is necessary.

DISCUSSION:

The speed with which the computer operates in this program gives the student an opportunity to make generalizations based upon many more observations then heretofore was possible. The question can be asked: " By what method does the computer find the prime factor?" A flow chart would be highly useful at this point, not only in developing the method for finding a prime factor, but also in understanding the mathematical logic behind this method.

```
THIS PROGRAM WILL GIVE YOU THE PRIME FACTORS OF ANY
WHOLE NUMBER.  IF YOU WISH TO STOP THE PROGRAM, ENTER A
ZERO FOR THE NUMBER.

WHAT IS THE NUMBER ? 105

 105            3  5  7

WHAT IS THE NUMBER ? 72

 72             2  2  2  3  3

WHAT IS THE NUMBER ? 89

 89           IS PRIME


WHAT IS THE NUMBER ? 47

 47           IS PRIME


WHAT IS THE NUMBER ? 155

 155            5  31

WHAT IS THE NUMBER ? 362

 362            2  181

WHAT IS THE NUMBER ? 0


READY
```

```
100 REM W. TEPPER WYANDANCH H.S.
105 REM  REVISED BY C.LOSIK  8-10-70
106 REM  M IS THE NUMBER, A(I) ARE ITS FACTORS
110 REM ADAPTATION OF TWO PROGRAMS
120 REM THIS PROGRAM FINDS THE PRIME FACTORS OF ANY GIVEN INTEGER
130 REM AND PRINTS PRIME IF IT HAS NO PROPER DIVISORS
140 DIM A(100)
150 LET C=0
160 PRINT "THIS PROGRAM WILL GIVE YOU THE PRIME FACTORS OF ANY"
170 PRINT "WHOLE NUMBER.  IF YOU WISH TO STOP THE PROGRAM, ENTER A"
172 PRINT "ZERO FOR THE NUMBER."
174 PRINT
180 PRINT "WHAT IS THE NUMBER ";
190 LET X=0
200 INPUT M
205 IF ABS(M-INT(M+.5))<.0001 THEN 210
206 PRINT "WHOLE NUMBERS ONLY, PLEASE."
207 GO TO 180
210 PRINT
215 IF M<=0 THEN 470
220 PRINT M,
230 LET I=1
240 LET I=I+1
245 IF I>M THEN 310
250 IF M/I<>INT(M/I) THEN 240
260 LET X=X+1
270 LET A(X)=I
280 LET M=M/I
300 GO TO 250
310 IF X=1 THEN 360
320 FOR L=1 TO X
330 PRINT A(L);
340 NEXT L
350 GO TO 370
360 PRINT "IS PRIME"
370 PRINT
380 PRINT
385 GO TO 180
400 INPUT B
410 IF B=1 THEN 180
420 IF B=0 THEN 470
430 PRINT " TYPE 1 OR 0 AS INSTRUCTED"
440 LET C=C+1
460 GO TO 400
470 END
```

DISCIPLINE MATHEMATICS 12, 13

SUBJECT ANALYTIC GEOMETRY

PROGRAM QUADRT

DESCRIPTION:

This program determines the nature of the graph of $Ax^2 + Bxy + Cy^2 + Dx + Ey + F = 0$, after the operator inputs the six constants (A, B, C, D, E, F). Limiting cases, such as a point or a line, are separated from the general cases so that the computer type-out gives the exact nature of the graph.

OBJECTIVE:

To permit exploration of the properties of the second-degree equation.

PRELIMINARY PREPARATION:

A. Student - should have a reasonable knowledge of conic sections, second - degree equations in two unknowns, invariant functions of the coefficients under transformations, etc.

B. Materials - An overhead projector along with a transparency of the flow chart would be desirable,

DISCUSSION:

Before running the program, the teacher should discuss the general form of a second-degree equation in two variables, the functions of the coefficients used in the program, and the implications of the flow chart.

The discussion of the flow chart for this program enhances the understanding of the problem.

The type-out serves as a check on students' efforts in identifying second-degree equations.

```
    THIS PROGRAM DETERMINES THE NATURE OF THE GRAPH OF:
         A*X↑2+B*X*Y+C*Y↑2+D*X+E*Y+F=0
    ENTER YOUR CONSTANTS IN THE ORDER LISTED ABOVE.

? 5,8,5,0,0,0
THE GRAPH OF YOUR EQUATION IS A SINGLE POINT.

ANOTHER RUN (1=YES, 0=NO) : ? 1

    ENTER YOUR CONSTANTS IN THE ORDER LISTED ABOVE.

? 5,8,5,0,0,36
THERE IS NO REAL LOCUS FOR YOUR EQUATION.

ANOTHER RUN (1=YES, 0=NO) : ? 1

    ENTER YOUR CONSTANTS INTTHE ORDER LISTED ABOVE.

? 0,5,0,0,0,9
THE GRAPH /F YOUR EQUATION IS A HYPERBOLA.

ANOTHER RUN (1=YES, 0=NO) : ? 0


READY
```

```
100 REM QUENTIN J.O'CONNOR, COMMACK H.S.NORTH, JULY 16, 1969
103 REM  REVISED BY C.LOSIK  8-7-70
105 REM  A,B,C,D,E,F ARE AS IN EQUATION
110 PRINT "     THIS PROGRAM DETERMINES THE NATURE OF THE GRAPH OF:"
120 PRINT "          A*X↑2+B*X*Y+C*Y↑2+D*X+E*Y+F=0"
130 PRINT "     ENTER YOUR CONSTANTS IN THE ORDER LISTED ABOVE."
140 PRINT " "
150 INPUT A,B,C,D,E,F
160 IF A*A+B*B+C*C+D*D+E*E+F*F>0 THEN 210
170 PRINT"     WITH ALL YOUR CONSTANTS EQUAL TO ZERO, ANY VALUES OF X"
180 PRINT"AND Y WILL SATISFY YOUR EQUATION. IN OTHER WORDS, YOUR"
190 PRINT"GRAPH IS A COMPLETE PLANE."
200 GO TO 550
210 IF A*A+B*B+C*C+D*D+E*E=0 THEN 500
220 IF A*A+B*B+C*C>0 THEN 240
230 GO TO 400
240 LET I=A+C
250 LET K=4*A*C-B*B
260 LET J=4*A*C+4*C*F+4*A*F-E*E-D*D-B*B
270 LET P=4*A*C*F+B*D*E-A*E*E-C*D*D-F*B*B
280 IF P=0 THEN 360
290 IF K=0 THEN 540
300 IF K<0 THEN 520
310 IF I*P>0 THEN 500
320 IF A<>C THEN 340
330 IF B=0 THEN 480
340 PRINT "THE GRAPH OF YOUR EQUATION IS AN ELLIPSE."
350 GO TO 550
360 IF K>0 THEN 460
370 IF K<0 THEN 440
380 IF J<0 THEN 420
390 IF J>0 THEN500
400 PRINT "THE GRAPH OF YOUR EQUATION IS A SINGLE STRAIGHT LINE."
410 GO TO 550
420 PRINT "THE GRAPH OF YOUR EQUATION CONSISTS OF 2 PARALLEL LINES."
430 GO TO 550
440 PRINT "THE GRAPH OF YOUR EQUATION CONSISTS OF 2 INTERSECTING LINES."
450 GO TO 550
460 PRINT "THE GRAPH OF YOUR EQUATION IS A SINGLE POINT."
470 GO TO 550
480 PRINT "THE GRAPH OF YOUR EQUATION IS A CIRCLE."
490 GO TO 550
500 PRINT "THERE IS NO REAL LOCUS FOR YOUR EQUATION."
510 GO TO 550
520 PRINT "THE GRAPH OF YOUR EQUATION IS A HYPERBOLA."
530 GO TO 550
540 PRINT "THE GRAPH OF YOUR EQUATION IS A PARABOLA."
550 PRINT
560 PRINT "ANOTHER RUN (1=YES, 0=NO) : ";
570 INPUT A
575 PRINT
580 IF A=1 THEN 130
590 IF A<>0 THEN 560
600 END
```

DISCIPLINE MATHEMATICS 9th YEAR

SUBJECT PROPORTIONS

PROGRAM NAME RATIO

DESCRIPTION:

This program solves a proportion of the type A/B = C/D. A, B, C, or D can be unknown.

OBJECTIVES:

A. To teach the student(s) the relationships in a proportion.

B. To aid in teaching the solution of proportions.

PRELIMINARY PREPARATION:

A. Student - no particular preparation necessary

B. Materials - see discussion

DISCUSSION:

The student is given the opportunity to see any number of solutions to proportions. The program then asks a series of questions designed to allow the student to discover that in a proportion, the product of the means equals the product of the extremes. The program can be used either with individual students or with an entire class depending upon the availability of equipment to display the output. The running time varies, depending upon the number of proportions you wish to solve. In 10 to 15 minutes, the program can be run with about 100 proportion problems. Included in this time is a built-in variable pause for observation of the tabulated results. Another value of using this program is that the teacher can easily handle numbers in proportions that heretofore were too difficult.

```
THIS PROGRAM SOLVES FOR THE UNKNOWN IN THE PROPORTION
A/B AS C/D. USE A ZERO AS A DUMMY VALUE FOR THE UNKNOWN.

HOW MANY PROPORTIONS DO YOU WISH TO SOLVE? 4
WHAT ARE THE VALUES FOR A,B,C,D? 3,5,8,9
YOU FORGOT TO INPUT A ZERO FOR YOUR
UNKNOWN.  TRY AGAIN.? 3,4,6,0
                                        3 / 4  AS  6 / 8
WHAT ARE THE VALUES FOR A,B,C,D? 1,10,0,50
                                        1 / 10  AS  5 / 50
WHAT ARE THE VALUES FOR A,B,C,D? 36,0,1,2
                                        36 / 72  AS  1 / 2
WHAT ARE THE VALUES FOR A,B,C,D? 0,45,3,5
                                        27 / 45  AS  3 / 5


TAKE A GOOD LOOK AT THE PROPORTIONS.  THE TWO MIDDLE
POSITION NUMBERS ARE CALLED THE 'MEANS', THE TWO
END POSITION NUMBERS ARE CALLED THE 'EXTREMES'.

LOOK AT THE 'MEANS' AND THE 'EXTREMES'  -    SEE IF
YOU CAN FIND SOME KIND OF RELATIONSHIP BETWEEN THEM.
WHEN YOU THINK YOU HAVE FOUND A RELATIONSHIP BETWEEN
THE 'MEANS' AND THE 'EXTREMES', TYPE 1 AND HIT THE RETURN KEY.
? 1


DID YOU SEE THAT IF YOU MULTIPLY THE 'MEANS'
AND MULTIPLY THE 'EXTREMES', THE PRODUCTS ARE EQUAL?

IN THE LAST PROPORTION  45 X 3 EQUALS 27 X 5
CHECK THE OTHERS, TOO.  WHEN YOU ARE READY TO CONTINUE,
TYPE 1 AND HIT THE RETURN KEY.
? 1


IF YOU WISH TO USE THIS PROGRAM AGAIN TYPE 1, IF NOT TYPE 0
? 0


READY
```

```
100 REM W. TEPPER WYANDANCH H.S.   -   MATHEMATICS
105 REM  REVISED BY C.LOSIK  5-5-70
106 REM  A/B = C/D, TOTALLY OBVIOUS.  ALSO USES GOSUB TO SIMULATE PAUSE
110 REM THIS PROGRAM SOLVES FOR THE UNKNOWN IN THE PROPORTION"
120 REM OF THE TYPE A/B AS C/D
130 PRINT "THIS PROGRAM SOLVES FOR THE UNKNOWN IN THE PROPORTION"
140 PRINT "A/B AS C/D. USE A ZERO AS A DUMMY VALUE FOR THE UNKNOWN."
150 PRINT
160 PRINT "HOW MANY PROPORTIONS DO YOU WISH TO SOLVE";
170 INPUT N
180 FOR K=1 TO N
190 PRINT "WHAT ARE THE VALUES FOR A,B,C,D";
200 INPUT A,B,C,D
210 IF A=0 THEN 270
220 IF B=0 THEN 290
230 IF C=0 THEN 310
240 IF D=0 THEN 330
250 PRINT "YOU FORGOT TO INPUT A ZERO FOR YOUR"
255 PRINT "UNKNOWN.  TRY AGAIN.";
260 GO TO 200
270 LET A=B*C/D
280 GO TO 340
290 LET B=A*D/C
300 GO TO 340
310 LET C=A*D/B
320 GO TO 340
330 LET D=B*C/A
340PRINT"                                                    ";A"/"B" AS "C"/"D
345 NEXT K
350 PRINT
360 PRINT
370 PRINT "TAKE A GOOD LOOK AT THE PROPORTIONS.  THE TWO MIDDLE"
380 PRINT "POSITION NUMBERS ARE CALLED THE 'MEANS'; THE TWO"
390 PRINT "END POSITION NUMBERS ARE CALLED THE 'EXTREMES'."
395 PRINT
400 PRINT "LOOK AT THE 'MEANS' AND THE 'EXTREMES'  -     SEE IF"
410 PRINT "YOU CAN FIND SOME KIND OF RELATIONSHIP BETWEEN THEM."
420 PRINT "WHEN YOU THINK YOU HAVE FOUND A RELATIONSHIP BETWEEN"
430 PRINT "THE 'MEANS' AND THE 'EXTREMES', ";
450 GO SUB 610
460 PRINT "DID YOU SEE THAT IF YOU MULTIPLY THE 'MEANS'"
470 PRINT "AND MULTIPLY THE 'EXTREMES', THE PRODUCTS ARE EQUAL?"
475 PRINT
480 PRINT "IN THE LAST PROPORTION "B"X"C"EQUALS" A"X"D
490 PRINT "CHECK THE OTHERS, TOO.  WHEN YOU ARE READY TO CONTINUE,"
510 GO SUB 610
540 PRINT
550 PRINT "IF YOU WISH TO USE THIS PROGRAM AGAIN TYPE 1, IF NOT TYPE 0"
560 INPUT X
570 IF X=1 THEN 200
580 IF X=0 THEN 640
590 PRINT "TYPE 1 OR 0 AS DIRECTED."
600 GO TO 560
610 PRINT "TYPE 1 AND HIT THE RETURN KEY."
620 INPUT X
623 IF X<>1 THEN 620
625 PRINT
627 PRINT
630 PRINT
635 RETURN
640 END
```

DISCIPLINE MATHEMATICS

SUBJECT QUADRATIC EQUATIONS

PROGRAM NAME ROOTS2

DESCRIPTION:

This program describes the nature of the roots of a quadratic equation, and finds the roots whether real or complex.

OBJECTIVES:

A. To familiarize the student with quadratic function.

B. To review and drill exercise... to study the nature of roots.

C. To emphasize that roots of $f(x) = 0$ are the same as x-intercepts of $f(x) = y$.

D. To impress the student with geometric interpretation(s) of the nature of roots.

E. To provide "lead-in" material for the introduction of further study of the real number line, the real cartesian plane, complex numbers, quadratic inequalities, etc.

PRELIMINARY PREPARATION:

A. Student - The teacher can use the program to introduce the students to the quadratic formula, to conclude discussion of the quadratic formula... or both.

B. Materials - none

DISCUSSION:

The program uses the "discriminant" to determine the nature of the roots of the quadratic equation. Regardless of the nature of the roots, the student is asked to graph $y = F(x)$, and to compare his graph with the kind of roots he finds for a specific $F(x) = 0$. He should be impressed with the picture; and he should understand (ultimately) the reasonableness and validity of the analytic methods presented in class.

```
THIS PROGRAM HANDLES ALL POSSIBLE CASES OF SOLUTION OF THE EQUATION :

     A * X ↑ 2  +  B * X  +  C  =  0

TYPE IN YOUR VALUES FOR A, B, AND C : ? 1,2,3

DISCRIMINANT IS LESS THAN ZERO, SO ROOTS ARE IMAGINARY.
THEY ARE OF THE FORM :  P+I*Q , P-I*Q , WHERE :
P =-1          Q = 1.414214

                                   ***


DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ? 1

TYPE IN YOUR VALUES FOR A, B, AND C : ? 1,7,3

DISCRIMINANT IS GREATER THAN ZERO, SO ROOTS ARE REAL.
ROOTS ARE X1 AND X2 .
X1 =-.4586187 X2 =-6.541381

                                   ***


DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ? 1

TYPE IN YOUR VALUES FOR A, B, AND C : ? 1,6,9

DISCRIMINANT IS EQUAL TO ZERO, SO ROOTS ARE EQUAL.  X =-3

                                   ***


DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ? 1

TYPE IN YOUR VALUES FOR A, B, AND C : ? 2,5,6

DISCRIMINANT IS GREATER THAN ZERO, SO ROOTS ARE REAL.
ROOTS ARE X1 AND X2
X1 =-1         X2 =-3

                                   ***


DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ? 0


READY
```

```
100 REM  THE ULTIMATE QUADRATIC SOLVER, UNTIL THE NEXT VERSION
110 REM  CHARLES LOSIK, PIB, 7/21/70, BASIC
120 PRINT "THIS PROGRAM HANDLES ALL POSSIBLE CASES OF SOLUTION OF";
125 PRINT " THE EQUATION :"
130 PRINT
140 PRINT "     A * X ↑ 2  +  B * X  +  C  =  0"
150 PRINT
160 PRINT "TYPE IN YOUR VALUES FOR A, B, AND C : ";
165 REM  INPUT VALUES FOR A,B,C
170 INPUT A,B,C
171 PRINT
175 REM  FOR ALL CASES, CHECK A=0. IF SO, THEN LINEARITY
180 IF A=0 THEN 802
185 REM  D IS THE DISCRIMINANT
190 LET D=B↑B-4*A*C
195 LET Z=2*A
200 IF D=0 THEN 710
210 IF D>0 THEN 610
300 REM  D<0, IMAGINARY RESULTS
310 PRINT "DISCRIMINANT IS LESS THAN ZERO, SO ROOTS ARE IMAGINARY."
320 PRINT "THEY ARE OF THE FORM :  P+I*Q , P-I*Q , WHERE I :"
330 PRINT "P ="-B/Z,"Q ="SQR(ABS(D))/Z
340 GO TO 900
600 REM  D>0, SO REAL ROOTS
610 PRINT "DISCRIMINANT IS GREATER THAN ZERO, SO ROOTS ARE REAL."
620 PRINT "ROOTS ARE X1 AND X2 ."
630 PRINT "X1 ="(-B+SQR(D))/Z,"X2 ="(-B-SQR(D))/Z
640 GO TO 900
700 REM  EQUAL ROOTS (D=0)
710 PRINT "DISCRIMINANT IS EQUAL TO ZERO, SO ROOTS ARE EQUAL.  X ="-B/Z
720 GO TO 900
800 REM  A=0, SO X=-C/B, UNLESS B=0
802 IF B<>0 THEN 810
803 IF C=0 THEN 807
804 PRINT "MEANINGLESS STATEMENT."
806 GO TO 900
807 PRINT "OK, ZERO = ZERO."
808 GO TO 900
810 PRINT "THE EQUATION IS LINEAR.  X ="-C/B
900 PRINT
901 PRINT TAB(30);"***"
905 PRINT
910 PRINT
920 PRINT "DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ";
930 INPUT Z
940 IF Z=1 THEN 150
950 IF Z<>0 THEN 920
999 END
```

DISCIPLINE MATHEMATICS - JR. HIGH

SUBJECT INTERSECTION AND UNION OF SETS

PROGRAM NAME SETS

DESCRIPTION:

This program finds the intersection and union of any two numerical sets.

OBJECTIVES:

A. To motivate students to find the union and intersection of any two sets.

B. To learn the logic involved in finding the union and intersection.

PRELIMINARY PREPARATION:

A. Student - no special preparation necessary.

B. Materials - see discussion

DISCUSSION:

This program may be used with individuals, small groups, or class-size groups. The elements of the two sets are entered as per instructions. Incidently, one or both of the sets may be empty. The computer then types back the elements in the union and intersection. The speed with which the computer operates enables the students to see a great many examples, giving them the opportunity to make discoveries about what is the union and what is an intersection of two sets. The teacher may use the flow chart that follows to explain the logic behind finding the union and intersection.

It is suggested that when used with large groups, a supplementary device be used to display output.

```
THIS PROGRAM FINDS THE UNION AND INTERSECTION OF ANY TWO
NUMERICAL SETS.

  HOW MANY ELEMENTS IN THE FIRST SET? 5

THESE ARE - (HIT THE RETURN KEY AFTER ENTERING EACH ELEMENT).
? 1
? 2
? 3
? 4
? 5

  HOW MANY ELEMENTS IN THE SECOND SET? 5

THESE ARE:
? 2
? 4
? 6
? 8
? 10


THE INTERSECTION CONTAINS  2  4
THE UNION CONTAINS 2  4  6  8  10  1  3  5

DO YOU WANT ANOTHER RUN (1=YES, 0=NO) : ? 1

  HOW MANY ELEMENTS IN THE FIRST SET? 8

THESE ARE - (HIT THE RETURN KEY AFTER ENTERING EACH ELEMENT).
? 1
? 2
? 3
? 4
? 6
? 8
? 10
? 12

  HOW MANY ELEMENTS IN THE SECOND SET? 10

THESE ARE:
? 1
? 2
? 3
? 4
? 5
? 6
? 7
? 8
? 9
? 10


THE INTERSECTION CONTAINS  1  2  3  4  6  8  10
THE UNION CONTAINS 1  2  3  4  5  6  7  8  9  10  12

DO YOU WANT ANOTHER RUN (1=YES, 0=NO) : ? 0


READY
```

```
100REM W. TEPPER, WYANDANCH HS, 7/29/69
101 REM  REVISED BY C.LOSIK  8-10-70
103 DIM A(30),B(30)
110 REM  UP TO 30 ELEMENTS PER SET ARE ALLOWED(UNLESS DIM IS CHANGED)
120PRINT"THIS PROGRAM FINDS THE UNION AND INTERSECTION OF ANY TWO"
130PRINT"NUMERICAL SETS."
140PRINT
150PRINT"  HOW MANY ELEMENTS IN THE FIRST SET";
160INPUT N
163 IF N=0 THEN 230
166 IF N=INT(N) THEN 170
167 PRINT "ONLY AN INTEGER NUMBER OF ELEMENTS IS POSSIBLE."
169 GO TO 140
170 IF N<=30 THEN 180
173 PRINT "THE MACHINE CANNOT HOLD MORE THAN 30 ELEMENTS.";
175 PRINT "  SEE YOUR TEACHER."
177 GO TO 690
180 IF N>0 THEN 189
183 PRINT "THERE CANNOT BE A NEGATIVE NUMBER OF ELEMENTS."
186 GO TO 140
189 PRINT
190PRINT"THESE ARE - (HIT THE RETURN KEY AFTER ENTERING EACH ELEMENT)."
200FOR K=1TON
210INPUTA(K)
220NEXT K
230PRINT
240 PRINT "  HOW MANY ELEMENTS IN THE SECOND SET";
250INPUT J
253 IF J=0 THEN 550
256 IF J=INT(J) THEN 260
257 PRINT "ONLY AN INTEGER NUMBER OF ELEMENTS IS POSSIBLE."
259 GO TO 230
260 IF J>30 THEN 173
270 IF J>0 THEN 279
273 PRINT "THERE CANNOT BE A NEGATIVE NUMBER OF ELEMENTS."
276 GO TO 230
279 PRINT
280 PRINT"THESE ARE:"
290 FOR K1=1TO J
300 INPUT B(K1)
310 NEXT K1
311 PRINT
312 PRINT
315 IF N<=0 THEN 640
320 PRINT "THE INTERSECTION CONTAINS ";
330 FOR K=1 TO N
340 FOR L=1 TO J
```

```
350 IF A(K)=B(L)THEN 380
360 NEXT L
370 GO TO 400
380 PRINT A(K);
390 LET X=X+1
400 NEXT K
410 IF X>0THEN 430
420 PRINT"  EMPTY SET....NO ELEMENTS"
430PRINT
440 PRINT"THE UNION CONTAINS";
450 FOR L=1 TO J
460 PRINT B(L);
470 NEXT L
480 FOR K=1 TO N
490 FOR L=1 TO J
500 IF A(K)=B(L)THEN 530
510 NEXT L
520 PRINT A(K);
530 NEXT K
540 GO TO 690
550 IF N<=0 THEN 620
560 PRINT "INTERSECTION IS EMPTY"
570 PRINT "UNION CONTAINS";
580 FOR K=1 TO N
590 PRINT A(K);
600 NEXT K
610 GO TO 690
620 PRINT "UNION AND INTERSECTION ARE EMPTY"
630 GO TO 690
640 PRINT "INTERSECTION IS EMPTY"
650 PRINT "UNION CONTAINS";
660 FOR K=1 TO J
670 PRINT B(K);
680 NEXT K
690 PRINT
700 PRINT
720 PRINT "DO YOU WANT ANOTHER RUN (1=YES, 0=NO) : ";
730 INPUT N
740 IF N=1 THEN 140
750 IF N<>0 THEN 720
760 END
```

DISCIPLINE ALGEBRA

SUBJECT SIMULTANEOUS EQUATIONS

PROGRAM NAME SIMEQN

DESCRIPTION:

This program finds the simultaneous solution set for sets of simultaneous linear equations (up to 10x10)

OBJECTIVES:

1. To eliminate the tedium of solution of sets of simultaneous equations.

2. To provide a means for checking solutions obtained by other means.

PRELIMINARY PREPARATIONS:

Presentation of concepts of simultaneous equations and methods for finding solutions.

```
THIS PROGRAM SOLVES ANY NUMBER OF SETS OF SIMULTANEOUS
EQUATIONS OF UP TO 10 EQUATIONS PER SET.  ENTER YOUR SETS
OF EQUATIONS IN DATA STATEMENTS IN LINES 700-800,
PRECEDED BY THE NUMBER OF EQUATIONS IN EACH SET.
EXAMPLE:  TO SOLVE THE SYSTEM
  1*X(1) + 2*X(2) = 3
  4*X(1) + 9*X(2) = 10
ENTER DATA AS FOLLOWS:
  700 DATA 2
  701 DATA 1,2,3
  702 DATA 4,9,10
THEN TYPE:
  1 GO TO 110
  RUN
THE COMPUTER WILL PRINT A MATRIX OF YOUR EQUATIONS, FOLLOWED
BY THE SOLUTION TO THE EQUATIONS.


READY

700 DATA 2
701 DATA 1,2,3
702 DATA 4,9,10
1 GO TO 110
RUN

 1             2             3
 4             9             10

X( 1 )=        7
X( 2 )=       -2




READY

700 DATA 2
701 DATA 3,2,16
702 DATA -6,-4,-32
1 GO TO 110
RUN

 3             2             16
-6            -4            -32

NO UNIQUE SOLUTION
```

```
READY

700 DATA 3
701 DATA 3,2,5,10
702 DATA -1,4,7,-21
703 DATA 1,1,-1,14
1 GO TO 110
RUN

 3              2              5             10
-1              4              7            -21
 1              1             -1             14

X( 1 )=          7.413044
X( 2 )=          2.956522
X( 3 )=         -3.630435


READY
```

```
10 REMARK  D.SOBIN, BKLYN POLY, 11-69
15 REM  REVISED BY C.LOSIK, 9-25-70
20 PRINT "THIS PROGRAM SOLVES ANY NUMBER OF SETS OF SIMULTANEOUS"
25 PRINT "EQUATIONS OF UP TO 10 EQUATIONS PER SET.  ENTER YOUR SETS"
30 PRINT "OF EQUATIONS IN DATA STATEMENTS IN LINES 700-800,"
35 PRINT "PRECEDED BY THE NUMBER OF EQUATIONS IN EACH SET."
40 PRINT "EXAMPLE:  TO SOLVE THE SYSTEM"
45 PRINT "  1*X(1) + 2*X(2) = 3"
50 PRINT "  4*X(1) + 9*X(2) = 10"
60 PRINT "ENTER DATA AS FOLLOWS:"
62 PRINT "  700 DATA 2"
64 PRINT "  701 DATA 1,2,3"
66 PRINT "  702 DATA 4,9,10"
70 PRINT "THEN TYPE:"
72 PRINT"  1 GO TO 110"
74 PRINT "  RUN"
80 PRINT "THE COMPUTER WILL PRINT A MATRIX OF YOUR EQUATIONS, FOLLOWED"
85 PRINT "BY THE SOLUTION TO THE EQUATIONS."
90 STOP
100 DIM E(10,11), X(10)
110 READ N
120 IF N=0 THEN 999
130 FOR I=1 TO N
140     FOR K=1 TO N+1
150 READ E(I,K)
155 PRINT E(I,K),
160     NEXT K
165 PRINT " "
170 NEXT I
185 REMARK  EVALUATE MATRIX
190 FOR J=1 TO N-1
200 IF E(J,J)=0 THEN 560
210 FOR I=J+1 TO N
220 LET Q=E(I,J)/E(J,J)
230 FOR K=J TO N+1
240 LET E(I,K)=E(I,K)-E(J,K)*Q
250 NEXT K
260 NEXT I
270 NEXT J
340 REMARK SOLVE FOR X(N)
350 IF E(N,N)=0 THEN 520
360 LET I=N+1
370 LET X(N)=E(N,I)/E(N,N)
380     FOR J=1 TO N-1
390     LET S=0
400       FOR K=1 TO J
410       LET S=S+E(N-J,I-K)*X(I-K)
```

```
420         NEXT K
430         LET X(N-J)=(E(N-J,I)-S)/E(N-J,N-J)
440       NEXT J
450 REMARK PRINT VALUES
455 PRINT
460 FOR J=1 TO N
470 PRINT "X("J")=",X(J)
480 NEXT J
500 GO TO 530
520 PRINT
525 PRINT "NO UNIQUE SOLUTION"
530 PRINT
535 PRINT
540 PRINT
550 GOTO 110
560 FOR T= J+1 TO N
570 IF E(T,J)<>0 THEN 600
580 NEXT T
590 GOTO 520
600 FOR C=J TO N+1
610 LET A=E(J,C)
620 LET E(J,C)=E(T,C)
630 LET E(T,C)=A
640 NEXT C
650 GOTO 210
801 DATA 0
999 END
```

DISCIPLINE CALCULUS - GRADE 13

SUBJECT TANGENT SLOPE FOR ANY FUNCTION

PROGRAM NAME SLOPE

DESCRIPTION:

This program considers a function which is differentiable at $x = a$, and at all points in the interval $[a, a+1]$. The value of the derivative at $x = a$ is approximated through secant slopes.

OBJECTIVES:

A. The preliminary discussion of the method whereby the machine solves the problem enhances the students' comprehension of the techniques. These techniques are then used in developing the analytic method for finding the slope of the tangent line.

B. The type-out of successive approximations to the tangent slope clarifies and dramatizes the nature of the limiting processes.

C. Time-saving factor through the elimination of lengthy computations.

PRELIMINARY PREPARATION:

Materials

The diagram below may be shown to the students on a blackboard, or an overhead projector, to explain the computations geometrically.



DISCUSSION:

The use of the computer and the attendant discussion of the program dramatically introduces the idea of differentiation.

```
                SECANT SLOPE OF A CURVE - THE DERIVATIVE

THIS PROGRAM CONSIDERS A FUNCTION OF X (Y=F(X)) WHICH IS
DIFFERENTIABLE AT X=A AND AT ALL POINTS IN THE INTERVAL
(A,A+1).  THE VALUE OF THE DERIVATIVE AT X=A IS
APPROXIMATED THROUGH SECANT SLOPES.

AFTER THE PROGRAM STOPS, TYPE IN THE FOLLOWING:
(END EACH LINE, INCLUDING 'RUN', WITH A 'CARRIAGE RETURN')

          1 GO TO 300
          300 DEF FNY(X)=....(YOUR FUNCTION OF X)....
          RUN

FOR EXAMPLE, TO FIND THE SLOPE OF THE EQUATION Y=X↑3
YOU WOULD TYPE AS FOLLOWS:

          1 GO TO 300
          300 DEF FNY(X)=X↑3
          RUN

YOU MIGHT TRY THAT AS YOUR FIRST RUN.
FOR SUBSEQUENT RUNS, YOU NEED ONLY CHANGE LINE 300 FOR
A NEW FUNCTION, FOLLOWED BY 'RUN'.


READY

1 GO TO 300
300 DEF FNY(X)=X↑3
RUN

FOR WHAT VALUE OF A IS THE SLOPE TO BE EVALUATED? 2

'CHANGE IN X' IS THE DISTANCE FROM 'A', AND 'CHANGE IN Y'
IS THE DISTANCE FROM 'F(A)' UPON WHICH THE SLOPE IS CALCULATED.


CHANGE IN X    CHANGE IN Y    SECANT SLOPE   % CHANGE IN SLOPE
------ -- -    ------ -- -    ------ -----   - ------ -- -----
1/ 1            19             19            NO PREVIOUS VALUE
1/ 2            7.625          15.25          19.73684
1/ 4            3.390625       13.5625        11.06557
1/ 8            1.595703       12.76562       5.875576
1/ 16           .7736816       12.37891       3.029376
1/ 32           .3808899       12.18848       1.53834
1/ 64           .1889687       12.09399       .7751783
1/ 128          .09411669      12.04694       .3891031
1/ 256          .04696667      12.02347       .1948049
1/ 512          .02346039      12.01172       .09771946
1/ 1024         .01172447      12.00586       .04878049
1/ 2048         5.860806E-3    12.00293       .02440215

*****

DO YOU WISH TO USE A DIFFERENT VALUE OF X (1-YES, 0-NO)? 0
TO CHANGE YOUR FUNCTION SEE THE INSTRUCTIONS.
IF YOU ARE FINISHED, TYPE '1', AND THE 'RETURN' KEY
AFTER THE PROGRAM STOPS.


READY

1
```

```
100 REM SECANT SLOPE OF A CURVE - Q. J. O'CONNOR  8-12-68
101 REM REVISED 8-7-70 (D. PESSEL)  (COMBINATION OF SLCUQ AND DIFFQ)
102 REM IMPORTANT VARIABLES:  S-SECANT SLOPE; P-PERCENT CHANGE;
103 REM D-CHANGE IN X; Y-CHANGE IN Y
105 LET S1=0
110 PRINT TAB(10);"SECANT SLOPE OF A CURVE - THE DERIVATIVE"
120 PRINT
130 PRINT "THIS PROGRAM CONSIDERS A FUNCTION OF X (Y=F(X)) WHICH IS"
131 PRINT "DIFFERENTIABLE AT X=A AND AT ALL POINTS IN THE INTERVAL"
132 PRINT "(A,A+1).  THE VALUE OF THE DERIVATIVE AT X=A IS"
133 PRINT "APPROXIMATED THROUGH SECANT SLOPES."
134 PRINT
139 PRINT "AFTER THE PROGRAM STOPS, TYPE IN THE FOLLOWING:"
140 PRINT "(END EACH LINE, INCLUDING 'RUN', WITH A 'CARRIAGE RETURN')"
141 PRINT
142 PRINT "          1 GO TO 300"
143 PRINT "          300 DEF FNY(X)=....(YOUR FUNCTION OF X)...."
145 PRINT "          RUN"
146 PRINT
147 PRINT "FOR EXAMPLE, TO FIND THE SLOPE OF THE EQUATION Y=X↑3"
148 PRINT "YOU WOULD TYPE AS FOLLOWS:"
149 PRINT
150 PRINT "          1 GO TO 300"
151 PRINT "          300 DEF FNY(X)=X↑3"
153 PRINT "          RUN"
154 PRINT
155 PRINT "YOU MIGHT TRY THAT AS YOUR FIRST RUN."
156 PRINT "FOR SUBSEQUENT RUNS, YOU NEED ONLY CHANGE LINE 300 FOR"
157 PRINT "A NEW FUNCTION, FOLLOWED BY 'RUN'."
160 STOP
290 REM CALCULATION OF SLOPE AND PRINTOUT
300 DEF FNY(X)=X↑3
305 PRINT "FOR WHAT VALUE OF A IS THE SLOPE TO BE EVALUATED";
306 INPUT A
310 PRINT
311 PRINT "'CHANGE IN X' IS THE DISTANCE FROM 'A', AND 'CHANGE IN Y'"
312 PRINT "IS THE DISTANCE FROM 'F(A)' UPON WHICH THE SLOPE IS CALCU";
313 PRINT "LATED."
316 PRINT
317 PRINT
320 PRINT "CHANGE IN X","CHANGE IN Y","SECANT SLOPE","% CHANGE IN SLOPE"
321 PRINT "------ -- -","------ -- -","------ -----","- ------ -- -----"
410 FOR N=0 TO 11
420 LET D=2↑N
430 LET Y=FNY(A+1/D)-FNY(A)
440 LET S=D*Y
444 IF S1>0 THEN 447
445 PRINT "1/"D,Y,S,"NO PREVIOUS VALUE"
446 GO TO 455
447 LET P=((ABS(S1-S))/S1)*100
450 PRINT "1/"D,Y,S,P
455 LET S1=S
460 NEXT N
470 PRINT
480 PRINT "*****"
490 PRINT
500 PRINT "DO YOU WISH TO USE A DIFFERENT VALUE OF X (1-YES, 0-NO)";
501 INPUT Q2
502 IF Q2>0 THEN 305
510 PRINT "TO CHANGE YOUR FUNCTION SEE THE INSTRUCTIONS."
520 PRINT "IF YOU ARE FINISHED, TYPE '1', AND THE 'RETURN' KEY"
530 PRINT "AFTER THE PROGRAM STOPS."
540 END
```

DISCIPLINE MATHEMATICS

SUBJECT ALGEBRA(9TH and 12TH GRADE)

PROGRAM NAME SQRT

DESCRIPTION:

This program finds the square root of counting numbers up to five decimal places.

OBJECTIVES:

A. To demonstrate and familiarize the students with square roots.

B. The method utilizes "pinching" $\sqrt{Z}$ between the endpoints of smaller and smaller domains.

PRELIMINARY PREPARATION:

A. Student - 1) The definition of square root as the inverse operation of squaring; and 2) Drill in estimating square roots to the nearest tenth, hundredth, etc.

B. Materials - none

DISCUSSION:

This program provides an "introduction to," and a "review of" evolution and involution. Limiting the neighborhood of $\sqrt{Z}$ to find successively closer approximations of the square root of a number, demonstrates to the student that he is able to determine the square root to any degree.

The program may be effectively utilized for introducing the limiting process.

```
PROGRAM FINDS SQUARE ROOT OF ANY POSITIVE NUMBER
BY 'PINCHING' IT WITHIN A SMALLER AND SMALLER INTERVAL.

WHAT IS THE NUMBER WHOSE SQUARE ROOT YOU SEEK? 54

LOWER LIMIT                                  UPPER LIMIT
----- -----                                  ----- -----
 0              < SQ.RT. OF 54 <              54
 5.4            < SQ.RT. OF 54 <              10.8
 7.02           < SQ.RT. OF 54 <              7.56
 7.344          < SQ.RT. OF 54 <              7.398
 7.344          < SQ.RT. OF 54 <              7.3494
 7.348319       < SQ.RT. OF 54 <              7.348859
 7.348427       < SQ.RT. OF 54 <              7.348481
 7.348465       < SQ.RT. OF 54 <              7.34847
 7.348469       < SQ.RT. OF 54 <              7.34847

APPROXIMATION NOW CORRECT TO AN ACCURACY OF 1.000000E-5
YOU MAY USE EITHER 7.348469 OR 7.34847 AS THE SQUARE ROOT OF 54

WANT TO TRY ANOTHER NUMBER (1=YES, 0=NO) : ? 1

WHAT IS THE NUMBER WHOSE SQUARE ROOT YOU SEEK? 39

LOWER LIMIT                                  UPPER LIMIT
----- -----                                  ----- -----
 0              < SQ.RT. OF 39 <              39
 3.9            < SQ.RT. OF 39 <              7.8
 6.24           < SQ.RT. OF 39 <              6.63
 6.24           < SQ.RT. OF 39 <              6.279
 6.2439         < SQ.RT. OF 39 <              6.2478
 6.24468        < SQ.RT. OF 39 <              6.24507
 6.244992       < SQ.RT. OF 39 <              6.245031
 6.244996       < SQ.RT. OF 39 <              6.245

APPROXIMATION NOW CORRECT TO AN ACCURACY OF 1.000000E-5
YOU MAY USE EITHER 6.244996 OR 6.245 AS THE SQUARE ROOT OF 39

WANT TO TRY ANOTHER NUMBER (1=YES, 0=NO) : ? 0

READY
```

```
100  REM  T. BURNS, JOHN GLENN HS,  8-6-69
110  REM  REVISED BY C.LOSIK   8-27-70
120  REM  A=LOWER LIMIT, B=UPPER LIMIT, Z=STEP IN INTERVAL
121  REM  E IS THE ACCURACY YOU DESIRE
125 LET E=.00001
130 PRINT "PROGRAM FINDS SQUARE ROOT OF ANY POSITIVE NUMBER"
140 PRINT "BY 'PINCHING' IT WITHIN A SMALLER AND SMALLER INTERVAL."
150 PRINT
160 PRINT
170 PRINT "WHAT IS THE NUMBER WHOSE SQUARE ROOT YOU SEEK";
180 INPUT Z
185 PRINT
190 IF Z>0 THEN 220
200 PRINT "YOUR NUMBER MUST BE POSITIVE !!!"
210 GO TO 160
220 PRINT
230 PRINT "LOWER LIMIT"," "," ","UPPER LIMIT"
235 PRINT "----- -----"," "," ","----- -----"
240 LET A=0
250 LET B=Z
260 LET S=(B-A)/10
270 PRINT A,"< SQ.RT. OF"Z"<",B
275 IF ABS(A*B-Z)<E THEN 360
280 FOR I=A TO B STEP S
290 IF Z<I*I THEN 310
300 NEXT I
301 LET B=B*10
302 GO TO 260
310 LET B=I
320 LET A=I-S
350 GO TO 260
360 PRINT
370 PRINT "APPROXIMATION NOW CORRECT TO AN ACCURACY OF"E
380 PRINT "YOU MAY USE EITHER"A"OR"B"AS THE SQUARE ROOT OF"Z
390 PRINT
400 PRINT
410 PRINT "WANT TO TRY ANOTHER NUMBER (1=YES, 0=NO) : ";
420 INPUT Z
430 IF Z=1 THEN 150
440 IF Z<>0 THEN 400
450 END
```

DISCIPLINE MATHEMATICS-TEACHER ASSISTANCE

SUBJECT ARITHMETIC MEAN (AVERAGE)

PROGRAM STATAL

DESCRIPTION:

This program finds the average (arithmetic mean), median, and standard deviation of up to one hundred numbers.

OBJECTIVES:

A. To familiarize the student with the concepts of arithmetic mean (average), median, and standard deviation of a group of numbers.

B. To impress him with the speed and accuracy of the computer as a calculating device.

C. To provide teachers with handy means of computing averages.

PRELIMINARY PREPARATION:

A. Student - "Arithmetic mean", "average", "median", and "standard deviation" must be well-defined.

B. Materials - None

DISCUSSION:

Given N terms, "A(1), A(2),..., A(N-1), A(N)", students will have learned the average of these N terms is "$\frac{A(1)+A(2)+...A(N-1)+A(N)}{N}$".

The program prints out the median value of the user's data when there is an odd number of data values. When there is an even number, the median value printed is the average between the N/2 and the (N+2)/2 terms.

The program serves as an excellent vehicle for drill in division and addition, and helps strengthen the concept of arithmetic mean (average).

This program is useful in demonstrating a simple "loop" routine for students interested in programming.

MEAN, MEDIAN, AND DEVIATION OF A SET OF NUMBERS.

ENTER YOUR NUMBERS IN DATA STATEMENTS ON LINES
1000 - 2000. FOR EXAMPLE, YOU MIGHT TYPE :

1000 DATA 1,2,3,4 ETC. (YOUR DATA GOES HERE)

WHEN YOUR DATA HAS BEEN ENTERED, TYPE :

1 GO TO 300
RUN

THEN RELAX WHILE THE MACHINE GRINDS OUT THE ANSWERS.

IF A 'SUBSCRIPT ERROR' APPEARS, INCREASE THE SIZE OF THE
ARRAY IN LINE 295.

WARNING -- THE NUMBER 9999 IS USED AS AN INTERNAL DATA
VALUE. IF THIS VALUE IS ONE OF YOUR DATA VALUES, SIMPLY
RE-TYPE LINES 999 AND 2001 WITH A COMMON DATA VALUE WHICH
YOU WILL NOT USE.

READY

```
1000 DATA 244,182,112,2,198,10,314,169,18,38
1 GO TO 300
RUN
```

THESE ARE YOUR NUMBERS :
244 182 112 2 198 10 314 169 18 38

THESE ARE YOUR NUMBERS (HIGHEST TO LOWEST) :
314 244 198 182 169 112 38 18 10 2

NUMBER OF VALUES IS 10
SUM OF THE VALUES IS 1287
THE MEAN VALUE IS 128.7
THE MEDIAN VALUE IS 140.5
THE STANDARD DEVIATION IS 209.5409

FOR ANOTHER RUN, RE-ENTER DATA ON LINES
1000 - 2000, TAKING CARE TO ELIMINATE OLD DATA
BY TYPING THOSE LINE NUMBERS WHICH YOU DO NOT USE AGAIN;
THEN TYPE 'RUN'.

READY

```
1

1000
```

```
100  REM  CHARLES M. LOSIK, BKLYN POLY, MEAN-MEDIAN-DEVIATION
110  REM  (7-66 IN FORTRAN II) ; (8-26-70 IN BASIC)
115 REM REVISED 9-24-70
120  REM  YOU PUT YOUR NUMBERS IN DATA STATEMENTS AND
130  REM  YOU GET WHAT YOU PAY FOR.
140 PRINT " ","MEAN, MEDIAN, AND DEVIATION OF A SET OF NUMBERS."
150 PRINT
160 PRINT " ENTER YOUR NUMBERS IN DATA STATEMENTS ON LINES"
170 PRINT " 1000 - 2000.  FOR EXAMPLE, YOU MIGHT TYPE :"
171 PRINT
172 PRINT " ","1000 DATA  1,2,3,4  ETC. (YOUR DATA GOES HERE)"
173 PRINT
174 PRINT " WHEN YOUR DATA HAS BEEN ENTERED,  TYPE :"
180 PRINT
190 PRINT " ","1 GO TO 300"
200 PRINT " ","RUN"
210 PRINT
220 PRINT " THEN RELAX WHILE THE MACHINE GRINDS OUT THE ANSWERS."
222 PRINT
225 PRINT " IF A 'SUBSCRIPT ERROR' APPEARS, INCREASE THE SIZE OF THE"
227 PRINT " ARRAY IN LINE 295."
230 PRINT
240  REM  A(I) ARE THE NUMBERS, S IS THEIR SUM,
250  REM  S2 IS THE SUM OF THEIR SQUARES.
260 REM
270 PRINT " WARNING -- THE NUMBER 9999 IS USED AS AN INTERNAL DATA"
275 PRINT " VALUE.  IF THIS VALUE IS ONE OF YOUR DATA VALUES, SIMPLY"
280 PRINT " RE-TYPE LINES 999 AND 2001 WITH A COMMON DATA VALUE WHICH"
285 PRINT " YOU WILL NOT USE."
290 STOP
295 DIM A(100)
300 PRINT
303 PRINT " THESE ARE YOUR NUMBERS :"
305 LET I=1
310 READ E
315 LET S=0
316 LET S2=0
320 READ A(I)
330 IF E = A(I) THEN 370
340 PRINT A(I) ;
345 LET S = S + A(I)
347 LET S2 = S2 + A(I) * A(I)
350 LET I = I + 1
360 GO TO 320
370 LET N = I - 1
380 PRINT
390 PRINT
399  REM  ***** BUBBLE SORT*****
400 PRINT " THESE ARE YOUR NUMBERS (HIGHEST TO LOWEST) :"
405 FOR I = 1 TO N - 1
```

```
 410 FOR J = I + 1 TO N
 420 IF A(I) > A(J) THEN 460
 430 LET T = A(I)
 440 LET A(I) = A(J)
 450 LET A(J) = T
 460 NEXT J
 465 PRINT A(I) ;
 470 NEXT I
 475 PRINT A(N)
 480 PRINT
 490 PRINT
 500 PRINT " NUMBER OF VALUES IS";N
 510 PRINT " SUM OF THE VALUES IS";S
 520 PRINT " THE MEAN VALUE IS" ; S / N
 530 PRINT " THE MEDIAN VALUE IS" ;
 540 IF N / 2 <> INT ( N / 2 ) THEN 570
 550 PRINT ( A(N/2) + A((N+2)/2))/2
 560 GO TO 600
 570 PRINT A((N+1)/2)
 600 PRINT " THE STANDARD DEVIATION IS" ; SQR ( N * S2 + S * S ) / N
 610 PRINT
 620 PRINT
 630 PRINT " FOR ANOTHER RUN, RE-ENTER DATA ON LINES"
 640 PRINT " 1000 - 2000, TAKING CARE TO ELIMINATE OLD DATA"
 642 PRINT " BY TYPING THOSE LINE NUMBERS WHICH YOU DO NOT USE AGAIN;"
 645 PRINT " THEN TYPE 'RUN'."
 650 STOP
 999 DATA 9999
2001 DATA 9999
2010 END

READY
```

DISCIPLINE MATHEMATICS, SOCIAL STUDIES

SUBJECT THE STOCK MARKET

PROGRAM NAME STOCK

DESCRIPTION:

This program simulates the stock market. Each student is given $10,000 with which he may buy and/or sell shares in five fictitious issues.

OBJECTIVES:

A. To give the student a simple understanding of the operations of the stock market.

B. To motivate the student to reinforce his basic arithmetic skills.

C. To give an example of the use of everyday mathematics and economics in everyday life.

PRELIMINARY PREPARATION:

A. Student - no special preparation

B. Materials - possibly graph paper

DISCUSSION:

This program can be used as a good motivation device in the teaching of basic stock-market concepts, and the basic mathematical skills involved. The computer starts each student with $10,000, and allows him to buy and/or sell shares. Precautionary tests are included for the student who tries to purchase more shares than he has money for, or to sell more shares than he actually owns. The program continues for as many trading days as the student desires.

The stock values rise and fall on a semi-random basis. On each trading day all stocks undergo a small random price change, a trend change (based on a random trend), and the possibility--on a random basis--of a large price change. The structure of the formula is:

new price=old price + (trend x old price) + (small random price change) + (possible large price change)

The trend is a random number between -.1 and +.1. It remains constant for a random number of days, at which time the trend is changed randomly. The trend affects all stocks equally, and attempts to simulate general market trends. The small random change ranges between -3 and +3 points. It occurs every day to every stock. The possible large price change is either +10 or -10 points. The + and - changes each occur at random day intervals, and to random stocks. That is, there may be no large change on some trading days, only a +10 change on others, a -10 change on still others, and both large and small changes on others. In all large-change cases, the change affects only one random stock when it occurs.

Because of the random generation of stock values and their fluctuations, the program does not exactly simulate the real market. It does, however, provide a simplified view of what does happen, and familiarizes the student with the basic functions involved. This should be explained to the students, along with some real causes of stock-market fluctuations.

Graph paper might be used to plot the daily stock values and the exchange average. In this way, the trend will become evident.

```
                    THE STOCK MARKET
DO YOU WANT THE INSTRUCTIONS (YES-TYPE 1, NO-TYPE 0)? 1


THIS PROGRAM PLAYS THE STOCK MARKET.  YOU WILL BE GIVEN
$10,000 AND MAY BUY OR SELL STOCKS.  THE STOCK PRICES WILL
BE GENERATED RANDOMLY AND THEREFORE THIS MODEL DOES NOT
REPRESENT EXACTLY WHAT HAPPENS ON THE EXCHANGE.  A TABLE
OF AVAILABLE STOCKS, THEIR PRICES, AND THE NUMBER OF SHARES
IN YOUR PORTFOLIO WILL BE PRINTED.  FOLLOWING THIS, THE
INITIALS OF EACH STOCK WILL BE PRINTED WITH A QUESTION
MARK.  HERE YOU INDICATE A TRANSACTION.  TO BUY A STOCK
TYPE +NNN, TO SELL A STOCK TYPE -NNN, WHERE NNN IS THE
NUMBER OF SHARES.  A BROKERAGE FEE OF 1% WILL BE CHARGED
ON ALL TRANSACTIONS.  NOTE THAT IF A STOCK'S VALUE DROPS
TO ZERO IT MAY REBOUND TO A POSITIVE VALUE AGAIN.  YOU
HAVE $10,000 TO INVEST.  USE INTEGERS FOR ALL YOUR INPUTS.
(NOTE:  TO GET A 'FEEL' FOR THE MARKET RUN FOR AT LEAST
10 DAYS)
-----GOOD LUCK!-----
```

```
STOCK                          INITIALS      PRICE/SHARE
INT. BALLISTIC MISSILES          IBM           85.75
RED CROSS OF AMERICA             RCA           85.5
LICHTENSTEIN, BUMRAP & JOKE      LBJ           155.25
AMERICAN BANKRUPT CO.            ABC           138
CENSURED BOOKS STORE             CBS           104.25

NEW YORK STOCK EXCHANGE AVERAGE:  113.75

TOTAL STOCK ASSETS ARE    $ 0
TOTAL CASH ASSETS ARE     $ 10000
TOTAL ASSETS ARE          $ 10000

WHAT IS YOUR TRANSACTION IN
IBM? 2
RCA? 3
LBJ? 1
ABC? 1
CBS? 1


**********  END OF DAY'S TRADING


STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             98.5           2              193            10.75
RCA             81             3              243           -4.5
LBJ             153.5          1              153.5         -1.75
ABC             135.5          1              135.5         -2.5
CBS             99             1              99            -5.25


NEW YORK STOCK EXCHANGE AVERAGE:  113.1    NET CHANGE: -.65

TOTAL STOCK ASSETS ARE    $ 824
TOTAL CASH ASSETS ARE     $ 9166.25
TOTAL ASSETS ARE          $ 9990.25

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 5
RCA? 1
LBJ? 1
ABC? 1
CBS? 0


**********  END OF DAY'S TRADING


STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             98.75          7              691.25         2.25
RCA             82.5           4              330            1.5
LBJ             154            2              308            .5
ABC             133.5          2              267           -2
CBS             102.75         1              102.75         3.75
```

```
NEW YORK STOCK EXCHANGE AVERAGE:  114.3    NET CHANGE:  1.2

TOTAL STOCK ASSETS ARE    $ 1699
TOTAL CASH ASSETS ARE     $ 8305.23
TOTAL ASSETS ARE          $ 10004.23

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 3
RCA? 2
LBJ? 5
ABC? -1
CBS? 3


**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             99.25          10             992.5          .5
RCA             82.25          6              493.5          -.25
LBJ             154.75         7              1083.25        .75
ABC             133.5          1              133.5          0
CBS             103.25         4              413            .5


NEW YORK STOCK EXCHANGE AVERAGE:  114.6    NET CHANGE:  .3

TOTAL STOCK ASSETS ARE    $ 3115.75
TOTAL CASH ASSETS ARE     $ 6882.5
TOTAL ASSETS ARE          $ 9998.25

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM?  5

RCA? 3
LBJ? 5
ABC? 3
CBS? 4


**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             96.75          15             1451.25        -2.5
RCA             80.5           9              724.5          -1.75
LBJ             150            12             1800           -4.75
ABC             132            4              528            -1.5
CBS             98.75          8              790            -4.5


NEW YORK STOCK EXCHANGE AVERAGE:  111.6    NET CHANGE: -3

TOTAL STOCK ASSETS ARE    $ 5293.75
TOTAL CASH ASSETS ARE     $ 4528.95
TOTAL ASSETS ARE          $ 9822.7
```

```
DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 0
RCA? -5
LBJ? -7
ABC? 0
CBS? -5


**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             96.75          15             1451.25        0
RCA             66.75          4              267            -13.75
LBJ             150.75         5              753.75         .75
ABC             132            4              528            0
CBS             95.75          3              287.25         -3


NEW YORK STOCK EXCHANGE AVERAGE:  108.4    NET CHANGE: -3.2

TOTAL STOCK ASSETS ARE    $ 3287.25
TOTAL CASH ASSETS ARE     $ 6455.74
TOTAL ASSETS ARE          $ 9742.99

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? -10
RCA? -2
LBJ? 2
ABC? 2
CBS? 0


**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             87.5           5              437.5          -9.25
RCA             58             2              116            -8.75
LBJ             135.25         7              946.75         -15.5
ABC             122.5          6              735            -9.5
CBS             98.75          3              296.25         3


NEW YORK STOCK EXCHANGE AVERAGE:  100.4    NET CHANGE: -8

TOTAL STOCK ASSETS ARE    $ 2531.5
TOTAL CASH ASSETS ARE     $ 6974.58
TOTAL ASSETS ARE          $ 9506.08

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? -4
RCA? -1
LBJ? -6
ABC? -8
CBS? -2
```

```
**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             80             1              80            -7.5
RCA             51             1              51            -7
LBJ             121.75         1              121.75        -13.5
ABC             109.5          4              438           -13
CBS             91.5           1              91.5          -7.25


NEW YORK STOCK EXCHANGE AVERAGE:  90.75    NET CHANGE: -9.65

TOTAL STOCK ASSETS ARE    $ 782.25
TOTAL CASH ASSETS ARE     $ 8619.96
TOTAL ASSETS ARE          $ 9402.21

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 0
RCA? 0
LBJ? 0
ABC? -3
CBS? 0

**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             77.5           1              77.5          -2.5
RCA             52.25          1              52.25          1.25
LBJ             119.25         1              119.25        -2.5
ABC             107            1              107           -2.5
CBS             92.25          1              92.25          .75


NEW YORK STOCK EXCHANGE AVERAGE:  89.65    NET CHANGE: -1.1

TOTAL STOCK ASSETS ARE    $ 446.25
TOTAL CASH ASSETS ARE     $ 8945.18
TOTAL ASSETS ARE          $ 9393.43

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 0
RCA? 0
LBJ? 0
ABC? 0
CBS? 10
```

```
**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             74.5           1              74.5          -3
RCA             54             1              54             1.75
LBJ             107            1              107           -12.25
ABC             108            1              108            1
CBS             90.75          11             998.25        -1.5


NEW YORK STOCK EXCHANGE AVERAGE:  86.85    NET CHANGE: -2.8

TOTAL STOCK ASSETS ARE    $ 1341.75
TOTAL CASH ASSETS ARE     $ 8013.46
TOTAL ASSETS ARE          $ 9355.21

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 5
RCA? 6
LBJ? 10
ABC? 10
CBS? 10

**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             72             6              432           -2.5
RCA             52.5           7              367.5         -1.5
LBJ             105            11             1155          -2
ABC             103.25         11             1135.75       -4.75
CBS             91.5           21             1921.5         .75


NEW YORK STOCK EXCHANGE AVERAGE:  84.85    NET CHANGE: -2

TOTAL STOCK ASSETS ARE    $ 5011.75
TOTAL CASH ASSETS ARE     $ 4221.92
TOTAL ASSETS ARE          $ 9233.67

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 0
HOPE YOU HAD FUN!!


READY
```

```
100 REM STOCK MARKET SIMULATION      -STOCK-
101 REM REVISED 8/18/70 (D. PESSEL, L. BRAUN, C. LOSIK)
102 REM IMP VRBLS: A-MRKT TRND SLP; B5-BRKRGE FEE; C-TTL CSH ASSTS;
103 REM C5-TTL CSH ASSTS (TEMP); C(I)-CHNG IN STK VAL; D-TTL ASSTS;
104 REM E1,E2-LRG CHNG MISC; I-STCK #; I1,I2-STCKS W LRG CHNG;
105 REM N1,N2-LRG CHNG DAY CNTS; P5-TTL DAYS PRCHSS; P(I)-PRTFL CNTNTS;
106 REM Q9-NEW CYCL?; S4-SGN OF A; S5-TTL DYS SLS; S(I)-VALUE/SHR;
107 REM T-TTL STCK ASSTS; T5-TTL VAL OF TRNSCTNS;
108 REM W3-LRG CHNG; X1-SMLL CHNG(<$1); Z4,Z5,Z6-NYSE AVE.; Z(I)-TRNSCTN
109  PRINT TAB(20);"THE STOCK MARKET"
110  DIM S(5),P(5),Z(5),C(5)
112 REM SLOPE OF MARKET TREND:A  (SAME FOR ALL STOCKS)
113  RANDOMIZE
114  LET A=INT((RND(X)/10)*100+.5)/100
115  LET T5=0
116  LET X9=0
117  LET N1=0
118  LET N2=0
119  LET E1=0
120  LET E2=0
121 REM INTRODUCTION
122  PRINT "DO YOU WANT THE INSTRUCTIONS (YES-TYPE 1, NO-TYPE 0)";
123  INPUT Z9
124  PRINT
125  PRINT
126  IF Z9<1 THEN 200
130  PRINT "THIS PROGRAM PLAYS THE STOCK MARKET.  YOU WILL BE GIVEN"
132  PRINT "$10,000 AND MAY BUY OR SELL STOCKS.  THE STOCK PRICES WILL"
134  PRINT "BE GENERATED RANDOMLY AND THEREFORE THIS MODEL DOES NOT"
135  PRINT "REPRESENT EXACTLY WHAT HAPPENS ON THE EXCHANGE.  A TABLE"
136  PRINT "OF AVAILABLE STOCKS, THEIR PRICES, AND THE NUMBER OF SHARES"
137  PRINT "IN YOUR PORTFOLIO WILL BE PRINTED.  FOLLOWING THIS, THE"
138  PRINT "INITIALS OF EACH STOCK WILL BE PRINTED WITH A QUESTION"
139  PRINT "MARK.  HERE YOU INDICATE A TRANSACTION.  TO BUY A STOCK"
140  PRINT "TYPE +NNN, TO SELL A STOCK TYPE -NNN, WHERE NNN IS THE"
141  PRINT "NUMBER OF SHARES.  A BROKERAGE FEE OF 1% WILL BE CHARGED"
142  PRINT "ON ALL TRANSACTIONS.  NOTE THAT IF A STOCK'S VALUE DROPS"
143  PRINT "TO ZERO IT MAY REBOUND TO A POSITIVE VALUE AGAIN.  YOU"
144  PRINT "HAVE $10,000 TO INVEST.  USE INTEGERS FOR ALL YOUR INPUTS."
145  PRINT "(NOTE:  TO GET A 'FEEL' FOR THE MARKET RUN FOR AT LEAST"
146  PRINT "10 DAYS)"
147  PRINT "-----GOOD LUCK!-----"
200 REM GENERATION OF STOCK TABLE; INPUT REQUESTS
210 REM INITIAL STOCK VALUES
220  LET S(1)=100
230  LET S(2)=85
240  LET S(3)=150
250  LET S(4)=140
260  LET S(5)=110
265 REM INITIAL T8 - # DAYS FOR FIRST TREND SLOPE (A)
266  LET T8=INT(4.99*RND(X)+1)
267 REM RANDOMIZE SIGN OF FIRST TREND SLOPE (A)
268  IF RND(X)>.5 THEN 270
269  LET A=-A
270 REM RANDOMIZE INITIAL VALUES
280  GOSUB 830
285 REM INITIAL PORTFOLIO CONTENTS
290  FOR I=1 TO 5
300  LET P(I)=0
305  LET Z(I)=0
310  NEXT I
320  PRINT
```

```
330  PRINT
333 REM INITIALIZE CASH ASSETS:C
335  LET C=10000
338 REM PRINT INITIAL PORTFOLIO
340  PRINT "STOCK","  ","INITIALS","PRICE/SHARE"
350  PRINT "INT. BALLISTIC MISSILES","  IBM",S(1)
352  PRINT "RED CROSS OF AMERICA","  RCA",S(2)
354  PRINT "LICHTENSTEIN, BUMRAP & JOKE","  LBJ",S(3)
356  PRINT "AMERICAN BANKRUPT CO.","  ABC",S(4)
358  PRINT "CENSURED BOOKS STORE","  CBS",S(5)
360  PRINT
361 REM NYSE AVERAGE:Z5; TEMP. VALUE:Z4; NET CHANGE:Z6
363  LET Z4=Z5
364  LET Z5=0
365  LET T=0
370  FOR I=1 TO 5
375  LET Z5=Z5+S(I)
380  LET T=T+S(I)*P(I)
390  NEXT I
391  LET Z5=INT(100*(Z5/5)+.5)/100
392  LET Z6=INT((Z5-Z4)*100+.5)/100
393 REM TOTAL ASSETS:D
394  LET D=T+C
395  IF X9>0 THEN 398
396  PRINT "NEW YORK STOCK EXCHANGE AVERAGE: "Z5
397  GO TO 399
398  PRINT "NEW YORK STOCK EXCHANGE AVERAGE: "Z5"   NET CHANGE: "Z6
399  PRINT
400  LET T=INT(100*T+.5)/100
401  PRINT "TOTAL STOCK ASSETS ARE   $";T
403  LET C=INT(100*C+.5)/100
405  PRINT "TOTAL CASH ASSETS ARE    $";C
407  LET D=INT(100*D+.5)/100
408  PRINT "TOTAL ASSETS ARE         $";D
410  PRINT
411  IF X9=0 THEN 416
412  PRINT "DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)";
413  INPUT Q9
414  IF Q9<1 THEN 998
416 REM INPUT TRANSACTIONS
420  PRINT "WHAT IS YOUR TRANSACTION IN"
430  PRINT "IBM";
440  INPUT Z(1)
450  PRINT "RCA";
460  INPUT Z(2)
470  PRINT "LBJ";
480  INPUT Z(3)
490  PRINT "ABC";
500  INPUT Z(4)
510  PRINT "CBS";
520  INPUT Z(5)
525  PRINT
530 REM TOTAL DAY'S PURCHASES IN $:P5
540  LET P5=0
550 REM TOTAL DAY'S SALES IN $:S5
560  LET S5=0
570  FOR I=1 TO 5
575  LET Z(I)=INT(Z(I)+.5)
580  IF Z(I)<=0 THEN 610
590  LET P5=P5+Z(I)*S(I)
600  GO TO 620
```

```
610  LET S5=S5-Z(I)*S(I)
612  IF -Z(I)<=P(I) THEN 620
614  PRINT "YOU HAVE OVERSOLD A STOCK; TRY AGAIN."
616  GO TO 420
620  NEXT I
622 REM TOTAL VALUE OF TRANSACTIONS:T5
625  LET T5=P5+S5
630 REM BROKERAGE FEE:B5
640  LET B5=INT(.01*T5*100+.5)/100
650 REM CASH ASSETS=OLD CASH ASSETS-TOTAL PURCHASES
652 REM -BROKERAGE FEES+TOTAL SALES:C5
654  LET C5=C-P5-B5+S5
656  IF C5>=0 THEN 674
658  PRINT "YOU HAVE USED $"-C5" MO
RE THEN YOU HAVE."
660  GO TO 420
674  LET C=C5
675 REM CALCULATE NEW PORTFOLIO
680  FOR I=1 TO 5
690  LET P(I)=P(I)+Z(I)
700  NEXT I
710 REM CALCULATE NEW STOCK VALUES
720  GOSUB 830
750 REM PRINT PORTFOLIO
751 REM BELL RINGING-DIFFERENT ON MANY COMPUTERS
752  FOR I=1 TO 20
753  PRINT CHR$(135);
754  NEXT I
755  PRINT
756  PRINT "**********  END OF DAY'S TRADING"
757  PRINT
758  PRINT
759  IF X9<1 THEN 769
769  PRINT "STOCK","PRICE/SHARE","HOLDINGS","VALUE","NET PRICE CHANGE"
770  PRINT "IBM", S(1), P(1), S(1)*P(1), C(1)
771  PRINT "RCA", S(2), P(2), S(2)*P(2), C(2)
772  PRINT "LBJ", S(3), P(3), S(3)*P(3), C(3)
773  PRINT "ABC", S(4), P(4), S(4)*P(4), C(4)
774  PRINT "CBS", S(5), P(5), S(5)*P(5), C(5)
775  LET X9=1
780  PRINT
790  PRINT
810  GO TO 360
829 REM NEW STOCK VALUES - SUBROUTINE
830 REM RANDOMLY PRODUCE NEW STOCK VALUES BASED ON PREVIOUS
831 REM DAY'S VALUES
832 REM N1,N2 ARE RANDOM NUMBERS OF DAYS WHICH RESPECTIVELY
833 REM DETERMINE WHEN STOCK I1 WILL INCREASE 10 PTS. AND STOCK
834 REM I2 WILL DECREASE 10 PTS.
840 REM IF N1 DAYS HAVE PASSED, PICK AN I1, SET E1, DETERMINE NEW N1
841  IF N1>0 THEN 850
845  LET I1=INT(4.99*RND(X)+1)
846  LET N1=INT(4.99*RND(X)+1)
847  LET E1=1
```

```
850 REM IF N2 DAYS HAVE PASSED, PICK AN I2, SET E2, DETERMINE NEW N2
851  IF N2>0 THEN 860
855  LET I2=INT(4.99*RND(X)+1)
856  LET N2=INT(4.99*RND(X)+1)
857  LET E2=1
860 REM DEDUCT ONE DAY FROM N1 AND N2
861  LET N1=N1-1
862  LET N2=N2-1
890 REM LOOP THROUGH ALL STOCKS
900  FOR I=1 TO 5
910  LET X1=RND(X)
915  IF X1>.25 THEN 920
916  LET X1=.25
917  GO TO 935
920  IF X1>.50 THEN 925
921  LET X1=.50
922  GO TO 935
925  IF X1>.75 THEN 930
926  LET X1=.75
927  GO TO 935
930  LET X1=0.0
931 REM BIG CHANGE CONSTANT:W3  (SET TO ZERO INITIALLY)
935  LET W3=0
936  IF E1<1 THEN 945
937  IF INT(I1+.5)<>INT(I+.5) THEN 945
938 REM ADD 10 PTS. TO THIS STOCK;  RESET E1
939  LET W3=10
943  LET E1=0
945  IF E2<1 THEN 955
947  IF INT(I2+.5)<>INT(I+.5) THEN 955
948 REM SUBTRACT 10 PTS. FROM THIS STOCK;  RESET E2
949  LET W3=W3-10
953  LET E2=0
954 REM C(I) IS CHANGE IN STOCK VALUE
955  LET C(I)=INT(A*S(I))+X1+INT(3-6*RND(X)+.5)+W3
956  LET C(I)=INT(100*C(I)+.5)/100
957  LET S(I)=S(I)+C(I)
960  IF S(I)>0 THEN 967
964  LET C(I)=0
965  LET S(I)=0
966  GO TO 970
967  LET S(I)=INT(100*S(I)+.5)/100
970  NEXT I
972 REM AFTER T8 DAYS RANDOMLY CHANGE TREND SIGN AND SLOPE
973  LET T8=T8-1
974  IF T8<1 THEN 985
980  RETURN
985 REM RANDOMLY CHANGE TREND SIGN AND SLOPE (A), AND DURATION OF
986 REM OF TREND (T8)
990  LET T8=INT(4.99*RND(X)+1)
992  LET A=INT((RND(X)/10)*100+.5)/100
993  LET S4=RND(X)
994  IF S4<=.5 THEN 997
995  LET A=-A
997  RETURN
998  PRINT "HOPE YOU HAD FUN!!"
999  END
```

DISCIPLINE CALCULUS-GRADE 13

SUBJECT AREA OF A SURFACE OF REVOLUTION

PROGRAM NAME SURFAR

DESCRIPTION:

This program approximates the area of a surface of revolution, by computing lateral areas of frustrums of cones of revolution.

OBJECTIVES:

A. The saving of time in computations.

B. The speedy demonstration of limiting processes.

C. The focusing of attention upon those processes needed to develop the analytic approach.

PRELIMINARY PREPARATION:

Before running this program, the lateral area of a frustrum of a cone should be discussed. Many students in the Advanced Placement Program have not taken a course in Solid Geometry and may be unfamiliar with the formula:

$$\text{Lateral Area} = \pi\, l\, (r_1 + r_2)$$



Frustrum of a Cone

Whether or not this formula is derived in class will depend on the amount of time available. Most likely it will merely be stated; students who have not taken Solid Geometry may be asked to look up the derivation on their own.

```
                AREA OF A SURFACE OF REVOLUTION

    THIS PROGRAM APPROXIMATES THE AREA OF A SURFACE OF
REVOLUTION BY COMPUTING LATERAL AREAS OF FRUSTUMS OF CONES
OF REVOLUTION.  TYPE IN YOUR FUNCTION OF X (Y=F(X)),
WHOSE GRAPH WILL BE ROTATED ABOUT THE X AXIS, AS FOLLOWS:

        1 GO TO 200
        300 DEF FNY(X)=...(YOUR FUNCTION OF X)...
        RUN

FOR EXAMPLE, TO USE THE FUNCTION Y=X↑2 YOU WOULD TYPE:

        1 GO TO 200
        300 DEF FNY(X)=X↑2
        RUN

YOU MIGHT TRY THAT AS YOUR FIRST RUN.
END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY.


READY

1 GO TO 200
300 DEF FNY(X)=X↑2
RUN

WHAT ARE THE ABSCISSAS OF THE END POINTS OF THE SECTION
TO BE CONSIDERED (SMALLER FIRST: P,Q)? -3,2

NUMBER OF       SUM OF                    % CHANGE
SUBINTERVALS    APPROXIMATING AREAS       IN SUM
------------    -------------------       --------
 1              288.7871                  NO PREVIOUS VALUE
 2              324.6229                   11.68411
 4              317.6819                   2.161263
 8              315.3346                   .7416313
 16             314.7434                   .1876635
 32             314.5933                   .04769154
 64             314.5557                   .01197374
 128            314.5461                   3.025796E-3

WOULD YOU LIKE TO TRY NEW END POINTS (1-YES, 0-NO)? 0
TO ENTER A NEW FUNCTION YOU NEED ONLY RETYPE LINE
300 AND 'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS.
IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY.


READY
```



```
1
```

```
100 REM AREA OF A SURFACE OF REVOLUTION, Q. J. O'CONNOR, 7/12/68
101 REM REVISED 8/21/70 (D. PESSEL)
105 PRINT TAB(17); "AREA OF A SURFACE OF REVOLUTION"
106 PRINT
110 PRINT "    THIS PROGRAM APPROXIMATES THE AREA OF A SURFACE OF"
120 PRINT "REVOLUTION BY COMPUTING LATERAL AREAS OF FRUSTUMS OF CONES"
130 PRINT "OF REVOLUTION.  TYPE IN YOUR FUNCTION OF X (Y=F(X)),"
131 PRINT "WHOSE GRAPH WILL BE ROTATED ABOUT THE X AXIS, AS FOLLOWS:"
150 PRINT
160 PRINT "        1 GO TO 200"
170 PRINT "        300 DEF FNY(X)=...(YOUR FUNCTION OF X)..."
180 PRINT "        RUN"
185 PRINT
186 PRINT "FOR EXAMPLE, TO USE THE FUNCTION Y=X↑2 YOU WOULD TYPE:"
187 PRINT
188 PRINT "        1 GO TO 200"
189 PRINT "        300 DEF FNY(X)=X↑2"
190 PRINT "        RUN"
191 PRINT
192 PRINT "YOU MIGHT TRY THAT AS YOUR FIRST RUN."
193 PRINT "END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY."
195 STOP
200 REM COMPUTATION SECTION OF PROGRAM
220 PRINT "WHAT ARE THE ABSCISSAS OF THE END POINTS OF THE SECTION"
230 PRINT "TO BE CONSIDERED (SMALLER FIRST: P,Q)";
240 INPUT P,Q
245 IF P<=Q THEN 250
246 PRINT "P CANNOT BE GREATER THAN Q!"
247 GO TO 220
250 PRINT
260 PRINT "NUMBER OF       SUM OF                          % CHANGE"
270 PRINT "SUBINTERVALS    APPROXIMATING AREAS             IN SUM"
280 PRINT "------------    -------------------             --------"
285 LET E1=0
300 DEF FNY(X)=X↑2
305 FOR N=1 TO 9
310 LET E=2↑(N-1)
320 LET H=(Q-P)/E
330 LET S=0
340 FOR I=0 TO (E-1)
350 LET G=FNY(P+I*H+H)+FNY(P+I*H)
360 LET M=FNY(P+I*H+H)-FNY(P+I*H)
370 LET L=3.14159*G*SQR(M*M+H*H)
380 LET S=S+L
390 NEXT I
395 IF S1=0 THEN 405
396 LET W=100*(ABS(S-S1))/((S+S1)/2)
399 IF S1=0 THEN 405
400 PRINT E,S," ",W
402 IF W<1E-2 THEN 420
404 GO TO 407
405 PRINT E,S," ","NO PREVIOUS VALUE"
407 LET S1=S
410 NEXT N
420 PRINT
430 PRINT "WOULD YOU LIKE TO TRY NEW END POINTS (1-YES, 0-NO)";
431 INPUT Q1
432 IF Q1>0 THEN 220
440 PRINT "TO ENTER A NEW FUNCTION YOU NEED ONLY RETYPE LINE"
450 PRINT "300 AND 'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS."
460 PRINT "IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY."
500 END
```

DISCIPLINE CALCULUS - GRADE 13

SUBJECT VOLUME OF ANY SOLID OF REVOLUTION, (ANALYTICALLY DEFINED)

PROGRAM NAME VOLSOL

DESCRIPTION:

Through the use of cylindrical discs, the program approximates the volume of a solid of revolution generated by rotating about the x-axis the area bounded by $y = f(x)$, the x-axis, and the vertical lines $x = a$ and $x = b$.

OBJECTIVES:

To help the student understand the analytic procedures and to appreciate the nature of the limiting process.

PRELIMINARY PREPARATION:

The class should be reminded of the formula for the volume of a cylinder, and the way in which a cylinder is generated by rotating a rectangle about one of its sides.

DISCUSSION:

It would be desirable to make use of an overhead projector transparency to display the cylindrical discs generated.



Approximation of Volume of Revolution by Cylindrical Discs

VOLUME OF A SOLID OF REVOLUTION

THIS PROGRAM USES CYLINDRICAL DISCS TO APPROXIMATE THE VOLUME OF A SOLID OF REVOLUTION. THE SOLID IS GENERATED BY ROTATING ABOUT THE X-AXIS THE AREA BOUNDED BY Y=F(X), THE LINES X=A AND X=B, AND THE X-AXIS.

TO INPUT YOUR FUNCTION OF X (Y=F(X)) TYPE AS FOLLOWS:

```
1 GO TO 200
220 DEF FNY(X)=....(YOUR FUNCTION OF X)....
RUN
```

FOR EXAMPLE, TO USE THE FUNCTION Y=X↑2 YOU WOULD TYPE:

```
1 GO TO 200
220 DEF FNY(X)=X↑2
RUN
```

YOU MIGHT TRY THAT AS YOUR FIRST EXAMPLE.
END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY.

READY

```
1 GO TO 200
220 DEF FNY(X)=X↑2
RUN
```

WHAT ARE YOUR VALUES FOR A AND B (SMALLER FIRST: A,B)? 0,5

NUMBER OF CYLINDERS	SUM OF CYLINDER VOLUMES	% CHANGE IN SUM
1	0	NO PREV. VALUE, OR IT WAS ZERO
2	306.7959	NO PREV. VALUE, OR IT WAS ZERO
4	939.5624	206.25
8	1400.955	49.10714
16	1669.476	19.16702
32	1813.291	8.614392
64	1887.594	4.097653
128	1925.344	1.999911
256	1944.369	.9881206
512	1953.918	.4911339

```
WOULD YOU LIKE TO TRY YOUR OWN 'NUMBER OF CYLINDERS' (1-YES, 0-NO)? 1
HOW MANY CYLINDERS WOULD YOU LIKE TO TRY? 700

FOR  700  CYLINDERS THE VOLUME IS  1956.487  .

WOULD YOU LIKE TO TRY AGAIN (1-YES, 0-NO)? 0

*****

WOULD YOU LIKE TO TRY NEW VALUES OF A AND B (1-YES, 0-NO)? 0
TO USE A NEW FUNCTION YOU NEED ONLY RETYPE LINE 220 AND
'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS.
IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY.


READY

1
```

```
100 REM VOLUME OF A SOLID OF REVOLUTION, Q. J. O'CONNOR, 8/1/68
101 REM REVISED 8/24/70 (D. PESSEL)
110 PRINT TAB(15);"VOLUME OF A SOLID OF REVOLUTION"
111 PRINT
115 PRINT"     THIS PROGRAM USES CYLINDRICAL DISCS TO APPROXIMATE"
117 PRINT"THE VOLUME OF A SOLID OF REVOLUTION.  THE SOLID IS GENE-"
120 PRINT"RATED BY ROTATING ABOUT THE X-AXIS THE AREA BOUNDED BY"
130 PRINT"Y=F(X), THE LINES X=A AND X=B, AND THE X-AXIS."
135 PRINT
140 PRINT"TO INPUT YOUR FUNCTION OF X (Y=F(X)) TYPE AS FOLLOWS:"
141 PRINT
145 PRINT"           1 GO TO 200"
150 PRINT"           220 DEF FNY(X)=....(YOUR FUNCTION OF X)...."
160 PRINT"           RUN"
161 PRINT
165 PRINT"FOR EXAMPLE, TO USE THE FUNCTION Y=X↑2 YOU WOULD TYPE:"
166 PRINT
167 PRINT"           1 GO TO 200"
168 PRINT"           220 DEF FNY(X)=X↑2"
169 PRINT"           RUN"
170 PRINT
175 PRINT"YOU MIGHT TRY THAT AS YOUR FIRST EXAMPLE."
176 PRINT"END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY."
180 STOP
199 PRINT
200 PRINT"WHAT ARE YOUR VALUES FOR A AND B (SMALLER FIRST: A,B)";
210 INPUT A,B
211 IF A<B THEN 214
212 PRINT"A MUST BE SMALLER THAN B!"
213 GO TO 200
214 PRINT
215 PRINT"NUMBER OF          SUM OF                    % CHANGE"
216 PRINT"CYLINDERS      CYLINDER VOLUMES               IN SUM"
217 PRINT"---------      ----------------          --------"
218 LET V1=0
220 DEF FNY(X)=X↑2
230 FOR N=1 TO 10
240 LET D=2↑(N-1)
250 LET H=(B-A)/D
260 LET V=0
270 FOR I=0 TO (D-1)
280 LET Y=FNY(A+I*H)
290 LET V=V+3.14159*Y*Y*H
300 NEXT I
305 IF V1=0 THEN 315
307 LET P=100*(ABS(V-V1))/V1
310 PRINT D,V," ",P
```

```
312 IF P<.5 THEN 330
313 GO TO 318
315 PRINT D,V,"    NO PREV. VALUE, OR IT WAS ZERO"
318 LET V1=V
320 NEXT N
330 PRINT
333 PRINT
334 PRINT"WOULD YOU LIKE TO TRY YOUR OWN 'NUMBER OF CYLINDERS' (1-YES";
335 PRINT", 0-NO)";
336 INPUT Q5
337 IF Q5<1 THEN 377
338 PRINT"HOW MANY CYLINDERS WOULD YOU LIKE TO TRY";
339 INPUT D1
340 IF D1>1 THEN 343
341 PRINT"NUMBER OF CYLINDERS MUST BE GREATER THAN ZERO!"
342 GO TO 338
343 IF D1<1000 THEN 347
344 PRINT"THIS IS A VERY LARGE NUMBER OF CYLINDERS AND MAY TAKE"
345 PRINT"A LONG TIME TO RUN."
347 LET V2=0
348 LET H1=(B-A)/D1
349 LET D1=INT(D1+.5)
350 FOR I=0 TO (D1-1)
352 LET Y1=FNY(A+I*H1)
354 LET V2=V2+3.14159*Y1*Y1*H1
356 NEXT I
358 PRINT
360 PRINT "FOR "D1" CYLINDERS THE VOLUME IS "V2" ."
362 PRINT
363 PRINT"WOULD YOU LIKE TO TRY AGAIN (1-YES, 0-NO)";
364 INPUT Q6
365 IF Q6>0 THEN 338
377 PRINT
378 PRINT"*****"
379 PRINT
380 PRINT"WOULD YOU LIKE TO TRY NEW VALUES OF A AND B (1-YES, 0-NO)";
382 INPUT Q1
384 IF Q1>0 THEN 199
386 PRINT"TO USE A NEW FUNCTION YOU NEED ONLY RETYPE LINE 220 AND"
388 PRINT"'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS."
390 PRINT"IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY."
500 END
```